

# XièXiè Programming Guide

*A beginner guidance for the XièXie programming language*

Lukas HERMANNS

Updated on April 11, 2015

# About the Author

My name is Lukas Hermanns (age-group 1990) and I started this project during my studies in 2014. By now I have over 12 years of experience in computer programming, started at the age of 12. I have been writing programs in Basic languages such as QBASIC, PUREBASIC, and BLITZ3D; in high level languages such as C, C++, C#, OBJECTIVE-C, and JAVA; but also in scripting languages such as JAVASCRIPT and PYTHON. I'm actually a preferred programmer in C++ (meanwhile C++11), low level stuff, and graphics programming with shading languages such as GLSL and HLSL.

The XiEXiE programming language is intended to be simple and not tuned for performance. It was originally designed to be used for scripting in video games, but can also be used for general purposes.

If you like, you can follow me on [Twitter](#), [YouTube](#), [GitHub](#), or [Bitbucket](#).

# ToDo List

The compiler, and partially the virtual machine, are not yet completed. Some rules and explanations in this report may change over time. Here is a rough ToDo-list:

PARSER	
Implementation of parser against grammar specification	done
CONTEXT ANALYZER	
Expression Type Check	done
Cast Type Check	incomplete
Automatic Type Deduction	done
Procedure Overloading	done
Procedure Overriding	done
Named Parameters	not yet implemented
Static and Non-Static Procedure Behavior	almost done
Class and Procedure Attributes	almost done
CODE GENERATOR	
Common CFG and TAC Generation from DAST	incomplete
CFG Simplification	almost done
If Statement	done
Switch Statement	incomplete
Return Statement	done
For Loop	done
Range-Based For Loop	done
Repeat Loop	done
While Loop	done
Do-While Loop	done
OPTIMIZER	
Constant Folding	done
Local Constant Propagation	almost done
Local Copy Propagation	not yet implemented
Local Variable Clean	almost done
Local Variable Reduction	incomplete
XASM BACKEND	
Final code generation for XASM	incomplete
Register Allocator	incomplete
Core Assembly File	incomplete

# Contents

<b>I</b>	<b>The XièXiè Programming Language</b>	<b>6</b>
<b>1</b>	<b>Introduction</b>	<b>7</b>
1.1	What is XièXiè?	7
1.2	Why is it called “XièXiè”?	7
1.3	Motivation	7
<b>2</b>	<b>Syntax</b>	<b>8</b>
2.1	Basics	8
2.1.1	Commentaries	8
2.1.2	Identifiers	8
2.1.3	Literals	9
2.2	Operators	9
2.3	Type Denoters	10
2.3.1	Built-in Types	10
2.3.2	Objects	10
2.3.3	Arrays	10
2.3.4	Automatic Type Deduction	10
2.4	Statements	11
2.4.1	Loop Statements	11
2.5	Expressions	13
2.5.1	Macros	13
<b>3</b>	<b>Classes</b>	<b>14</b>
3.1	Getting Started	14
3.2	Declaration Rules for Classes	15
3.3	Procedures	16
3.3.1	Procedure Signature	16
3.3.2	Procedure Overloading	16
3.3.3	Procedure Overriding	16
3.4	Attributes	17
3.5	Anonymous Classes	17
3.6	Class Visibilities	18
<b>4</b>	<b>Modules</b>	<b>19</b>
4.1	Using Modules	19
4.2	Creating Modules	19
<b>II</b>	<b>Developer Tools</b>	<b>21</b>
<b>5</b>	<b>Compiler</b>	<b>22</b>
5.1	Command Line Tool	22
5.1.1	Output	22
5.1.2	Script	22
<b>6</b>	<b>Virtual Machine</b>	<b>24</b>
6.1	Executing Programs	24

<b>III</b>	<b>Low-Level Programming</b>	<b>25</b>
<b>7</b>	<b>Virtual Assembler</b>	<b>26</b>
7.1	Instruction Set . . . . .	26
7.1.1	Registers . . . . .	26
7.1.2	OpCodes . . . . .	26
7.1.3	Memory Alignment . . . . .	27
7.2	Procedure Calling . . . . .	28
7.2.1	Calling Convention . . . . .	28
7.2.2	Intrinsics . . . . .	28

## **Part I**

# **The XIÈXIE Programming Language**

# Chapter 1

## Introduction

First of all, “xièxie” is the chinese word for “thanks” or “thank you” (see [hantrainerpro.com](http://hantrainerpro.com)) and is roughly pronounced “Sh-eh sh-eh”.

The design of the XièXie programming language is overall influenced by `JAVA`, `C++`, and `PYTHON`.

### 1.1 What is XièXie?

The XièXie programming language is a high-level, object-oriented, scripting language with compiler and virtual machine. The XièXie compiler (XXC) translates the XièXie code (XX) to a virtual assembler (XASM), then assembles it to XièXie byte code (XBC), which can then be interpreted by the XièXie virtual machine (XVM).

### 1.2 Why is it called “XièXie”?

Some years, before I started with the development of this compiler, I already had some ideas about a name for it — at this time the compiler was intended to be a cross compiler, i.e. to compile the XièXie code to `C++`. The first idea I had in mind, was to call it “C+=2” or “C++++” because it should be a more comfortable and easier `C++` variant. But this name looked and sounded strange. Another idea was to call it “C power of C” (in mathematical notation  $C^C$ ). But that still was not what I was looking for. Then I remembered me to the Chinese word “Xièxie” which means “Thanks” in English and it sounds a little similar to “CC” which could be seen as a shortcut for  $C^C$ . That’s the XièXie compiler’s story about its name.

Or to make a long story short: I saw this word on a napkin in a Chinese restaurant and thought to my self: That’s it! :-)

### 1.3 Motivation

There are many great programming languages out there. Some produce faster code than others, but some are simpler and have a better learning curve. Using a new language doesn’t mean to give up a previous one, because every language has its own domain. For example, the performance of interpreted languages such as `PYTHON` is totally sufficient for many applications. We don’t need maximal performance for a script which does some socket connection or text processing for instance. But I would not write the compiler and interpreter for `PYTHON` in a scripted language. There are also many ways to combine several languages: an interpreter could be used for scripting in video games for instance. But the game engine is written in `C++`. This is how it is done in `UNITY3D` (see [www.unity3d.com](http://www.unity3d.com)). They use `MONO` as compiler and interpreter framework for `C#`. But the engine itself is written in `C++`.

Now XièXie is aimed to be used for small scripting purposes, with a gentle learning curve. The great thing about its interpreter is, that it is very tiny and can easily be integrated into existing `C` or `C++` code. This virtual machine only consists of a single code file (written in `C99`) and can simply be included into any `C99` compliant project.

# Chapter 2

## Syntax

We start out with the syntax.

### 2.1 Basics

#### 2.1.1 Commentaries

Commentaries are a fundamental part of programming languages and they are nearly identical to those in `JAVA`:

```
// Single-line comment

/* Single-line comment */

/*
Multi-line comment
*/
```

Although they are very similar to the commentaries in `JAVA`, nested multi-line comments are allowed as well:

```
/*
Outer comment
/* Nested comment */
*/
```

#### 2.1.2 Identifiers

Identifiers (for variables, classes, etc.) must only contain alpha-numeric characters and the underscore. They must also begin with a letter or an underscore. But there is a small exception: they must not begin with `__xx__`, because all identifiers with this prefix are reserved for internal use of the compiler.

Valid identifiers are:

- `_name_`
- `FooBar`
- `number_of_wheels`
- `Customer01`
- ...

Invalid identifiers are:

- `naïve`
- `Foo.Bar`
- `number-of-wheels`
- `3over2`
- ...



### 2.1.3 Literals

These are the kinds of literals:

- Boolean Literal: `true`, `false`
- Integer Literal (Binary): e.g. `0b11001`, `0b0000`
- Integer Literal (Octal): e.g. `0o24`, `0o01234567`
- Integer Literal (Decimal): e.g. `3`, `12`, `999`, `1234567890`
- Integer Literal (Hexa-Decimal): e.g. `0xff`, `0x00`, `0xaB29`
- Float Literal: e.g. `0.0`, `3.5`, `12.482`
- String Literal: e.g. `"Foo Bar"`, `"Hello, World"`, `"\n\t"`, `"1st fragment"_"2nd fragment"`
- Verbatim String Literal: e.g. `@"\home\test"`, `@"a ""b"" c"`

Integer and float literals may optionally contain the single quotation mark as digit separator, for better readability. If it's used, all separators must satisfy the following rules:

1. For decimal literals, the separators must be **three steps apart** from each other, beginning at the dot.  
Example: `12'345`, `3.141'592'654`, or `-27'836.283'74`.
2. For non-decimal literals, the separators must be **four steps apart** from each other, beginning at the dot.  
Example: `0xff0'214b`, `0b100'1101'1011`, `0o1234'5670`
3. A separator must not appear at the beginning or the end of the literal.  
Counterexample: `'123'456'`.
4. No valid separator must be omitted.  
Counterexample: `1234'567'890`.

## 2.2 Operators

The most operators as in JAVA or C++ are also available in XIE:

```
/* Arithmetic Operators */
a + b    // Addition
a - b    // Substration
a * b    // Multiplication
a / b    // Division
a % b    // Modulo
a << b   // Left Shift
a >> b   // Right Shift
-a       // Negate

/* Bitwise Operators */
a & b    // Bitwise AND
a | b    // Bitwise OR
a ^ b    // Bitwise XOR
~a       // Bitwise NOT

/* Boolean Operators */
not a    // Logic NOT
a and b  // Logic AND
a or b   // Logic OR

/* Relation Operators */
a = b    // Equality
a != b   // Inequality
a < b    // Less
a <= b   // Less Or Equal
a > b    // Greater
a >= b   // Greater Or Equal
```

As the interested reader may have noticed, the *equality operator* is different to that in the most languages. This is because the *copy assignment operators* is `:=` and not `=`.

```
x := 5           // ok, set x to 5
a, b, c := 4    // ok, set a, b, and c to 4
a := b := 3     // error, b := 3 is not an expression
```

In the above example `a := b := 3` is invalid, because on the right hand side of the first `:=` must be an expression. But per definition in XiEXIE `b := 3` is not an expression, but a statement! The variable list assignment (`a, b, c := ...`) is a comfort functionality, which is only supported for the copy assignment.

The modify-assign operators are available as well:

```
a += b
a -= b
a *= b
a /= b
a %= b
a <<= b
a >>= b
a &= b
a |= b
a ^= b
```

However, they are also not allowed inside another expression.

## 2.3 Type Denoters

### 2.3.1 Built-in Types

There are only the following three built-in data types:

```
bool // Boolean type; can be 'true' or 'false'
int  // 32-bit signed integral type
float // 32-bit floating-point type
```

### 2.3.2 Objects

Types for class objects (more about classes in chapter 3) are written as follows:

```
// Empty string
String s

// List of strings
String s1 := "Hello, World", s2 := "Foo", s3 := "Bar"
```

### 2.3.3 Arrays

The only generic way for lists are the built-in arrays:

```
// Declare array objects with initializer lists
int[] intArray := { 1, 2, 3 }
float[][] floatArrayArray := { { 0.0, 1.5 }, { 3.5, 1.23 } }
String[] stringArray := { "a", "b", "c" }

// Access array elements
String s1 := stringArray[0]
String s2 := stringArray[intArray[0]]
float[] floatArray := floatArrayArray[1]
```

### 2.3.4 Automatic Type Deduction

Whereas automatic type deduction in C++11 is a very extensive language feature, in XiEXIE it can be summarized in this section. There are two keywords for automatic type deduction: `var` and `const`. As the name implies `var` denotes a variable type and `const` denotes a constant type. The latter type is the only way to define constants in XiEXIE. Here are a few examples:

```
var i := 1           // i is from type 'int'
var f := 3.5         // f is from type 'float'
var s := "."          // s is from type 'String'
var a := { 5 }       // a is from type 'array of int'
```

```

var aa := {           // aa is from type 'array of array of String'
  { "test" },
  { "a", "b" }
}

const ci := 5         // ci is a constnat int with value 5
const cj := ci*2      // cj is a constant int with value 10
const cf := 3.14      // cf is a constant float
const cb := ci > cj   // cb is a constant bool with value 'false'

```

## 2.4 Statements

Unlike C++ and Java, there are no semicolons in XiEXIE to terminate statements. Only the regular for-statement has two semicolons, to separate the initializer statement, the condition expression, and the increment statement. This means the compiler (or rather the *parser*, which reads the source code) always knows when a statement or expression is complete. But it also means that you — the programmer — can do weird things with this syntax. Consider the following code sample:

```

int a:=3
-4
,b:=-
5+2 int c

```

This is valid XiEXIE code and it contains only two statements! If we write it in a more common convention, it may look like this:

```

int a := 3-4,
    b := -5+2
int c

```

Hence, the readability of your code is up to you and your programming style :-). The only language I've worked with, which forces you to practice better readability is Python. Actually a great principle, but with XiEXIE you have complete freedom.

The absence of statement terminators is the reason for the *double paren* syntax of attributes:

```
[[attribute]]
```

To understand why this is the case, let's assume attributes are written with a single paren and consider the following class declaration:

```

class Widget {
  const c := 0 // Initialize member constant 'c' with 0
  int v := c // Initialize member variable 'v' with constant 'c'
  [entry]    // Mark next procedure as the main entry point with attribute 'entry'
  static void main() {}
}

```

Now this doesn't seem very complex. But the parser runs into trouble when reading [entry]. This is because the parser reads it as follows:

```
int v := c[entry]
```

But c is not an array. This is why attributes are written with a double paren, because array accesses never begin with '['. They may end with ']', but this is not important for the parser.

### 2.4.1 Loop Statements

It follows several examples of loop statements.

#### for Loop

```

// This regular for loop iterates 'i' from 0 to 9 (similar to Java)
for int i := 0 ; i < 10 ; i++ {
  // Infinite loop (also similar to Java)
  for ;; {
    if i >= 0 {
      // Break infinite loop
      break
    }
  }
}

```

```

    }

    // Inner iteration variable 'j' is implicit initialized to 0.0
    float j
    for ; j < 3.5 ; {
        j += 0.5
    }
}

```

## Ranged Based for Loop

```

// Print numbers 0, 1, 2, 3, and 4
for i : 0 .. 4 {
    // Print value of 'i' (this 'i' is not mutable!)
    print(i)
}

// Print numbers 10, 7, 4, 1, -2, -5, and -8
for i : 10 .. -10 -> 3 {
    print(i)
}

```

## foreach Loop

```

// Iterate over array with elements 1, 2, and 3
foreach i : { 1, 2, 3 } {
    print(i)
}

// Iterate over a 'superList' from type (array of arrays of strings)
String[][] superList := { { "a", "b" }, { "c" } }
foreach list : superList {
    // Iterate over all elements in the current sub list
    foreach str : list {
        // Do something with this string
        print(str)
    }
}

```

## repeat Loop

```

// Repeat for unconditional iterations
// (This is internally a "ForEver"-loop)
repeat {
    // Condition to break the loop
    if magicFunction() {
        // Break loop
        break
    }
}

// Do something 10 times (with invisible index variable)
// (This is internally a "ForRange"-loop)
repeat 10 {
    doSomething()
}

```

## while Loop

```

// Regular while loop
while magicFunction() {
    doSomething()
}

```

## do/while Loop

```
// Regular do-while loop
do {
    doSomething()
} while magicFunction()
```

## 2.5 Expressions

### 2.5.1 Macros

XiEXIE does not support the declaration of macros. However, there are a few built-in macros, which represent additional reserved identifiers:

- **\_\_FILE\_\_** String which contains the current filename.
- **\_\_CLASS\_\_** String which contains the current class name.
- **\_\_PROC\_\_** String which contains the current procedure name.
- **\_\_LINE\_\_** Integer which contains the current line number.
- **\_\_DATE\_\_** String which contains the current date and time.
- **\_\_VERSION\_\_** Integer which contains the compiler version number (e.g. 200 for version "2.00").

# Chapter 3

## Classes

A XièXiè program can only consist of imports, modules, and class declarations. And classes can only be defined in the global scope. That means every procedure must be defined inside a class and inner classes are currently not supported.

### 3.1 Getting Started

To get started, take a look at the following example program which prints the classical phrase “Hello, World!” onto the standard output:

```
1 // XieXie Hello World Program
2 import System
3 class HelloWorld {
4     static void main() {
5         System.out.WriteLine("Hello, World!")
6     }
7 }
```

This merely writes the line “Hello, World!” to the standard output. Let’s take a closer look at each line.

Line 2 imports the “System.xx” file from the XièXiè standard library:

```
import System // either this ...
import "System.xx" // ... or this
```

If the imported file is in another directory, the string version of `import` is the only choice. This can also be omitted if the file is added to the compilation process. The files for the classes `Object`, `String`, `Array`, and `Intrinsics` are always implicit imported, because they are ‘internal’ classes the compiler knows generally. Note that verbatim strings are allowed wherever string literals are allowed, i.e. the following example is valid XièXiè code:

```
import "C:\\Program_Files\\Test1.xx" // either this ...
import @"C:\\Program_Files\\Test1.xx" // ... or this
```

The `import` keyword is different to that in `JAVA` and also different to the `#include` directive in `C++`. Although it takes a filename as parameter (like `C++`’s `#include`), it does not *include* the file in place. Whenever an `import` is read by the *parser*, the filename is added to the set of import files. After all source files have been read, which were passed as input to the compiler, all import files will be read next. This will be repeated until no new files are added to the set. Consider this is a *set* of files, i.e. several `import` commands may occur with the same filename, but it will be read only once. This is why the above sample is valid XièXiè code. It also means that recursive imports are allowed as well:

```
// File1.xx
import "File2.xx"
```

```
// File2.xx
import "File1.xx"
```

Line 3 declares the class `HelloWorld` which implicit inherits from the base class `Object`, like it is done in `JAVA`:

```
3 class HelloWorld { /* ... */ }
```

To inherit from other classes, just write a colon and the identifier of the base class:

```
class SubClass : BaseClass { /* ... */ }
```

There is no multiple inheritance like in C++ or interfaces like in Java!

The next line declares the procedure main.

```
4 static void main() {
```

This is the main entry point for the program. There can only be one main entry point, but there are several possible signatures for this procedure:

```
// No return value, no arguments
static void main() { /* ... */ }

// No return value, arguments
static void main(String[] args) { /* ... */ }

// Return value, no arguments
static int main() { /* ... */ }

// Return value, arguments
static int main(String[] args) { /* ... */ }
```

The last line of code prints the message to the standard output:

```
5 System.out.writeln("Hello, World!")
```

System is a class from the standard `XiXiE` library, out is a *static* member from the type 'OutputStream', and writeln is a function which takes a string as input.

## 3.2 Declaration Rules for Classes

In `XiXiE` there is no need for *forward declarations*. Everything can be declared in the respective scope and is accessible throughout the entire program (except private scope). This is why the following code is valid:

```
// First declare sub class
class SubClass : BaseClass { /* ... */ }

// Then declare base class
class BaseClass { /* ... */ }
```

The same applies for procedure declarations:

```
class B {
    static void procB1() {
        A.procA() // no forward declaration required ...
        B.procB2() // ... same here
    }
    static void procB2() { /* ... */ }
}
class A {
    static void procA() { /* ... */ }
}
```

This works because the *context analyzer* of the compiler works in several phases:

1. Class symbols are registered in global scope.
2. Class signatures are analyzed (attributes and base class).
3. Class inheritance is verified (check for cycles).
4. Member procedures are registered in respective class scope.
5. Member variables are registered in respective class scope.
6. Run-time type information (RTTI) for the entire class hierarchy is generated.
7. Procedure code is analyzed.

## 3.3 Procedures

In XièXiè we talk about *procedures* (somethings called *methods*), because in the strict sense *functions* have no side effects. Functions have only input parameters which are calculated to a result. But in XièXiè every procedure can have side effects, meaning that they can modify the program state (with static variables for instance).

### 3.3.1 Procedure Signature

A *procedure signature* is the extended identification of a procedure beyond its identifier string. The signature consists of the identifier string, its parameter list, and the procedure's return type. However, the return type is never used for identification.

### 3.3.2 Procedure Overloading

XièXiè supports overloading of procedures. This means the same identifier can be used several times for procedures inside a class declaration (including its inheritance hierarchy). This requires that the following rules are satisfied:

All procedures with the same identifier inside a class declaration can be distinguished by their parameter count or parameter types.

Here is an example of procedure overloading.

```
class Widget {
  int f() {
    return 1
  }
  int f(int x) {
    return x*2
  }
  float f(float x) {
    return x*3.0
  }
  void caller() {
    int a := f() // a is 1
    int b := f(1) // b is 2
    float c := f(2.0) // c is 6.0
  }
}
```

When overloading procedures, try to avoid ambiguities with default arguments. Adding default arguments to all parameters of all overloaded procedures is allowed, but procedure calls may be ambiguous for the compiler:

```
class Widget {
  int f() { /* ... */ }
  int f(int x := 0) { /* ... */ }
  float f(float x := 0.0) { /* ... */ }

  void caller() {
    int a := f() // error, could be f(), f(int x := 0), or f(float x := 0.0)
    int b := f(1) // ok, argument is from type 'int'
    float c := f(2.0) // ok, argument is from type 'float'
  }
}
```

### 3.3.3 Procedure Overriding

XièXiè supports overriding of procedures. This means the same procedure signature can be used inside a class and its base class. The procedure calls of overloaded and overridden procedures require that the following rules are satisfied:

If a procedure with identifier *P* is declared inside a class *C*, a procedure with the same identifier is declared in its base class *B* but with another signature, and another procedure inside class *C* calls the procedure *P* from class *B*, then the identifier *super* must be specified in front of the call.

To better understand this awkward definition, take a look at this example:



```

class B {
    int f(int x, int y) {
        return 1
    }
    int g() {
        return 2
    }
}
class S : B {
    int f(int x) {
        return 3
    }
    void caller() {
        int a := f(0)           // equivalent to 'this.f(0)'
        int b := super.f(0, 0)  // 'super' is required, due to overloaded procedure 'f'
        int c := g()           // no need for 'super', because 'g' is not overloaded
    }
}

```

### 3.4 Attributes

There are several attributes for class-, procedure-, and variable declarations:

```

// Class A is marked as 'deprecated'.
[[deprecated]]
class A {
    // Declare some procedure
    void p() {}
}

// Class B is marked as 'deprecated' with a hint.
// Since B is deprecated as well, the deprecation of A is ignored here.
[[deprecated("hint...")]]
class B {
    // Mark 'p' to override 'A.p'. If 'p' does not match the signature
    // of procedure 'A.p', the compiler will throw an error message.
    [[override]]
    void p() {}
}

// Class C is 'final', i.e. no class can inherit from C.
// Because A is deprecated but not C, the compiler will throw a warning.
[[final]]
class C : A {}

```

Here is a summary of all attributes:

- **deprecated**, **deprecated(String)** Marks a class, procedure, or member-variable as deprecated.
- **override** Overrides a procedure from a class in its inheritance hierarchy.
- **final** Makes a class final, which disables to inherit from this class.

### 3.5 Anonymous Classes

XiEXIE supports the declaration of anonymous classes. This can be quite comfortable when a class needs to be instantiated only once:

```

class Hello {
    // No code block after procedure declaration inside
    // a non-extern class implies an abstract procedure
    String say()
}

class EnglishHello : Hello {
    [[override]]
    String say() { return "Hello" }
}

class Main {
    static void main() {

```

```

// Allocate class "EnglishHello"
var englishHello = new EnglishHello()

// Allocate anonymous class (internal name "__xx__AnonymousClass0")
var germanHello = new Hello() {
    [[override]]
    String say() { return "Hallo" }
}
}
}

```

Now it should be clear why identifiers with the prefix "\_\_xx\_\_" are reserved.

## 3.6 Class Visibilities

Like in C++ and JAVA, XIE provides visibilities for all members within a class. The syntax for these visibilities are a mixture of C++ and JAVA. You can declare the visibility for each class member like in JAVA and/or for a whole bunch of members like in C++:

```

class C {
public:
    int    publicVar0
    float  publicVar1
    private int    privateVar0
    String publicVar2
    protected int  protectedVar0
    private float[] privateVar1
    int[] publicVar3
private:
    int    privateVar2
    public bool    publicVar4
    bool[] privateVar3
}

```

Unlike in common object-oriented languages, the default visibility in XIE is public. This is because there are only classes and no structures like in C++. So a small class, which only has variables and acts like a data structure can be written with as less effort as possible:

```

// Examples of simple data structures
class Vector3 { float x, y, z }
class FileHeader {
    int magicNumber
    int formatVersion
}

```

# Chapter 4

## Modules

A *module* consists of a shared library (\*.dll on WINDOWS and \*.so on GNU/LINUX) and a XIE module file. They are similar to PYTHON modules.

### 4.1 Using Modules

Module declarations are similar to external classes. But they can only have static procedures. Here is an example:

```
module MyModule {
    static void doSomething(int x)
}
```

Using this in your XIE code will then look like this:

```
import MyModule
class MyClass {
    static void main() {
        MyModule.doSomething(42)
    }
}
```

### 4.2 Creating Modules

Modules must be written in plain C or C++. A minimal module would consist of three files. Supposed our example module is named “MyModule” and we work on WINDOWS we would have these files:

- **MyModule.c** Module code written in C.
- **MyModule.dll/ .so** Shared library which runs the module code.
- **MyModule.xx** XIE code file which declares the module.

The C code file must implement the following function interfaces to be a valid XIE module:

```
// Returns the number of module procedures.
int xx_module_proc_count();

// Returns the procedure for the respective index.
XVM_INVOCATION_PROC xx_module_fetch_proc(int index);

// Returns the procedure identifier for the respective index.
const char* xx_module_fetch_ident(int index);
```

Our simple module example from above could be implemented as follows:

```
#include "../xx_module.h"

// This is our module procedure "soSomething".
// All module procedures must have this interface.
void doSomething(xvm_env env) {
    // Get 1st parameter from XVM environment
    int x = xvm_param_int(env, 1);
```

```
// Do something with 'x'
printf("input parameter 'x' is %i\n", i);

// Pop arguments from stack (1 for the single parameter 'x')
xvm_return_void(env, 1);
}

// Implement the module interface
static xvm_invocation procList[] = {
    { "doSomething", doSomething },
};

XVM_IMPLEMENT_MODULE_INTERFACE(procList);
```

# **Part II**

## **Developer Tools**

# Chapter 5

## Compiler

### 5.1 Command Line Tool

The compiler can be used as command line tool. The appropriate program is named `xxc`. Enter `xxc help` in a command line to see the manual pages.

In contrast to most other command line tools, the commands for `xxc` don't have the `'-'` prefix. Instead the command options use this prefix. Here is an example:

```
xxc C -f FILE1 -f FILE2 -O
```

The above command line uses the `C` command (also `compile`) with the flags `'-f'` and `'-O'`.

#### 5.1.1 Output

The compiler tries to show you where an error or warning occurred. If possible, it is highlighted with a *line marker*. Consider the following code sample:

```
class Foo {  
    void Bar() {  
        int x := y  
    }  
}
```

Now consider the compiler processes this code with the following command line:


```
xxc C -f Core.xx -f Foo.xx
```

Then the compiler indicates that the variable `'y'` was not declared (see Figure 5.1). As you can see, some errors may produce multiple outputs. In this case, `'y'` is undeclared which produces two error outputs here:

1. The identifier `'y'` is undeclared, so the appearance in the initializer expression is invalid.
2. The type of the initializer expression for the declaration of `'x'` can not be deduced, so the type check fails.


#### 5.1.2 Script

If `xxc` has no arguments, it will switch to the *immediate script* mode. This allows the user to immediately script `XIEXIE` code inside the command line (see Figure 5.2).



```
Windows PowerShell
parse file "Core.xx" ...
parse file ".\CompilerOutputTest1.xx" ...
context analysis ...
2 error(s), 0 warning(s):
-----
context error (3:11 - 3:12) -- undeclared identifier "y"
>>   int x := y
>>   ~
context error (3:6 - 3:7) -- invalid type of initializer expression in variable declaration
>>   int x := y
>>   ~
```


(a) Output in POWERSHELL on WINDOWS 7



```
Code: bash - Konsole
Datei Bearbeiten Ansicht Lesezeichen Einstellungen Hilfe
parse file "Core.xx" ...
parse file "CompilerOutputTest1.xx" ...
context analysis ...
2 error(s), 0 warning(s):
-----
context error (3:11 - 3:12) -- undeclared identifier "y"
>>   int x := y
>>   ~
context error (3:6 - 3:7) -- invalid type of initializer expression in variable declaration
>>   int x := y
>>   ~
```

(b) Output in TERMINAL on KUBUNTU GNU/LINUX

Figure 5.1: Compiler output for the above code sample.



```
Windows PowerShell
PS C:\Users\Lukas Hermanns> xxc
XieXie Compiler 2.0 Alpha
Copyright (c) 2014-2015 Lukas Hermanns
01  for i : { 42, 7, -3, 9 } {
02      print(i)
03  }_
```

Figure 5.2: Compiler immediate script mode.

## Chapter 6

# Virtual Machine

The XièXiè VIRTUAL MACHINE (XVM) is a separated program (named `xvm`), written in pure C99.

### 6.1 Executing Programs

To execute (or run) a virtual program, just enter the filename of an `*.xbc` file into the `xvm`:

```
xvm HelloWorld.xbc
```



**Part III**

**Low-Level Programming**

# Chapter 7

## Virtual Assembler

The XiEXIE Virtual Assembler (XASM) is the low level language which is used as interface to the xvm. It's a very low level language, with similarities to ARM® Assembler.

### 7.1 Instruction Set

#### 7.1.1 Registers

The xvm is a register machine, i.e. it uses registers as operands for its instructions instead of the stack. This commonly increases performance but is a little more tricky to use, when you run out of registers. Each register index inside an instruction is stored with 4 bits, thus there are 16 registers and this is the list:

1. **\$r0** General Purpose Register 0.
- ⋮
25. **\$r24** General Purpose Register 24.
26. **\$ar** Argument Return.
27. **\$xr** Extended Register.
28. **\$gp** Global Pointer.
29. **\$cf** Conditional Flag.
30. **\$lb** Local Base Pointer.
31. **\$sp** Stack Pointer.
32. **\$pc** Program Counter.

All pointer registers and the program counter (i.e. \$gp, \$lb, \$sp, and \$pc) are all 'real' pointers to the memory in your operating system.

#### 7.1.2 OpCodes

Each instruction stores its mnemonic (Greek 'memory'  $\triangleq$  instruction identifier) in the first 6 bits. This provides a maximum of 64 instructions. However, only 52 mnemonics are currently in use.

3-Register Instruction OpCodes (00....)						
Mnemonic	OpCode 31.....26	Dest. 25.....21	LSource 20.....16	RSource 15.....11	Unused 10.....0	Description
AND	0 0 0 0 0 1	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Bitwise AND (D := L & R).
OR	0 0 0 0 1 0	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Bitwise OR (D := L   R).
XOR	0 0 0 0 1 1	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Bitwise XOR (D := L ^ R).
ADD	0 0 0 1 0 0	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Integer add. (D := L + R).
SUB	0 0 0 1 0 1	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Integer sub. (D := L - R).
MUL	0 0 0 1 1 0	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Integer mul. (D := L × R).
DIV	0 0 0 1 1 1	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Integer div. (D := L ÷ R).
MOD	0 0 1 0 0 0	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Modulo (D := L % R).
SLL	0 0 1 0 0 1	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Shift left (D := L « R).
SLR	0 0 1 0 1 0	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Shift right (D := L » R).
ADDF	0 0 1 0 1 1	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Float add. (D := L + R).
SUBF	0 0 1 1 0 0	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Float sub. (D := L - R).
MULF	0 0 1 1 0 1	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Float mul. (D := L × R).
DIVF	0 0 1 1 1 0	D D D D D	L L L L L	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0	Float div. (D := L ÷ R).

2-Register Instruction OpCodes (01....)					
Mnemonic	OpCode 31.....26	Dest. 25.....21	Source 20.....16	Unused or Value 15.....0	Description
MOV	0 1 0 0 0 0	D D D D D	S S S S S	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Move (D := S).
NOT	0 1 0 0 0 1	D D D D D	S S S S S	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Bitwise NOT (D := ~S).
FTI	0 1 0 0 1 0	D D D D D	S S S S S	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Float to integer (D := (int)S).
ITF	0 1 0 0 1 1	D D D D D	S S S S S	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Integer to float (D := (float)S).
AND	0 1 0 1 0 0	D D D D D	S S S S S	V V V V V V V V V V V V V V V V	Bitwise AND (D := S & V).
OR	0 1 0 1 0 1	D D D D D	S S S S S	V V V V V V V V V V V V V V V V	Bitwise OR (D := S   V).
XOR	0 1 0 1 1 0	D D D D D	S S S S S	V V V V V V V V V V V V V V V V	Bitwise XOR (D := S ^ V).
ADD	0 1 0 1 1 1	D D D D D	S S S S S	V V V V V V V V V V V V V V V V	Modulo (D := S % V).
SUB	0 1 1 0 0 0	D D D D D	S S S S S	V V V V V V V V V V V V V V V V	Shift left (D := S « V).
MUL	0 1 1 0 0 1	D D D D D	S S S S S	V V V V V V V V V V V V V V V V	Shift right (D := S » V).
DIV	0 1 1 0 1 0	D D D D D	S S S S S	V V V V V V V V V V V V V V V V	Integer add. (D := S + V).
MOD	0 1 1 0 1 1	D D D D D	S S S S S	V V V V V V V V V V V V V V V V	Integer sub. (D := S - V).
SLL	0 1 1 1 0 0	D D D D D	S S S S S	V V V V V V V V V V V V V V V V	Integer mul. (D := S × V).
SLR	0 1 1 1 0 1	D D D D D	S S S S S	V V V V V V V V V V V V V V V V	Integer div. (D := S ÷ V).
CMP	0 1 1 1 1 0	X X X X X	Y Y Y Y Y	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Integer comparison (→ %cf).
CMPF	0 1 1 1 1 1	X X X X X	Y Y Y Y Y	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Float comparison (→ %cf).

1-Register Instruction OpCodes (100...)				
Mnemonic	OpCode 31.....26	Reg. 25.....21	Unused, Value, or Offset 20.....0	Description
PUSH	1 0 0 0 0 0	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Push register onto stack.
POP	1 0 0 0 0 1	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Pop register from stack.
INC	1 0 0 0 1 0	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Increment integer (D++).
DEC	1 0 0 0 1 1	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Decrement integer (D--).
MOV	1 0 0 1 0 0	R R R R R	V V V V V V V V V V V V V V V V	Move integer (D := V).
LDA	1 0 0 1 0 1	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Load address from program (D := ByteCode + R <sub>word</sub> + O <sub>byte</sub> ).

Jump Instruction OpCodes (101...)				
Mnemonic	OpCode 31.....26	Reg. 25.....21	Offset 20.....0	Description
JMP	1 0 1 0 0 0	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump.
JE	1 0 1 0 0 1	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if equal.
JNE	1 0 1 0 1 0	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if not-equal.
JG	1 0 1 0 1 1	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if greater.
JL	1 0 1 1 0 0	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if less.
JGE	1 0 1 1 0 1	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if greater or equal.
JLE	1 0 1 1 1 0	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if less or equal.
CALL	1 0 1 1 1 1	R R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Push dynamic link (\$1b and \$pc) onto stack; Set \$1b to new stack frame; Jump to address.

Load/Store Instruction OpCodes (1100..)					
Mnemonic	OpCode 31.....26	Reg. 25.....21	Addr. 20.....16	Offset 15.....0	Description
LDB	1 1 0 0 0 0	R R R R R	A A A A A	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Load byte from memory.
STB	1 1 0 0 0 1	R R R R R	A A A A A	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Store byte to memory.
LDW	1 1 0 0 1 0	R R R R R	A A A A A	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Load word from memory.
STW	1 1 0 0 1 1	R R R R R	A A A A A	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Store word to memory.

Special-1 Instruction OpCodes			
Mnemonic	OpCode 31.....26	Unused or Value 25.....0	Description
STOP	0 0 0 0 0 0	0 0	Stop program execution.
PUSH	1 1 1 0 0 0	V V	Push integer onto stack.
INVK	1 1 1 0 0 1	V V	Invoke external procedure.
INSC	1 1 1 0 1 0	V V	Call intrinsic.

Special-2 Instruction OpCodes				
Mnemonic	OpCode 31.....26	Result Size 25.....16	Argument Size 15.....0	Description
RET	1 1 1 0 1 1	R R R R R R R R R R	A A A A A A A A A A A A A A A A	Pop R words from stack and buffer them; Pop current stack frame; Pop A words from stack; Push stored R words back onto stack; Restore dynamic link (\$1b and \$pc)

## 7.1.3 Memory Alignment

Some instructions use *byte* alignment and some others use *word* alignment.

## 7.2 Procedure Calling

### 7.2.1 Calling Convention

In XASM, the calling convention provides that the *callee* removes the procedure arguments from the stack. After any procedure call, the *dynamic link* is stored at the beginning of the *stack frame* and reserves the first 8 bytes.

#### dynamic link

Consists of two words: 32-bit value of the \$1b register *before* the procedure call and 32-bit value of the \$pc register *before* the procedure call.

Local variables can be store by increasing the *stack pointer* (\$sp) and load/store instructions at the *local base pointer* (\$1b) plus 8 bytes (after the *dynamic link*). Here is an example:

```
1 ; void proc() { ... }
2 proc:
3     ADD $sp, $sp, 8    ; Allocate memory on the stack for two words (2*4 bytes)
4     MOV $r0, 42        ; Store value 42 in register $r0
5     STW $r0, ($1b) 8   ; Store register $r0 in local scope at first position (after dynamic link)
6     ADD $r0, $r0, 10   ; Increment value of register $r0
7     STW $r0, ($1b) 12  ; Store register $r0 in local scope at second position
8     ...
```

The arguments for a procedure call are meant to be pushed onto the stack from *right-to-left*. In this way the arguments can be accessed from *left-to-right*. Here is an example:

```
1 ; int proc(int x, int y) { return x+y }
2 proc:
3     LDW $r0, ($1b) -4  ; Fetch first argument (x) from stack
4     LDW $r1, ($1b) -8  ; Fetch second argument (y) from stack
5     ADD $ar, $r0, $r1  ; Calculate ar := x + y
6     RET 2              ; Return: $ar and pop 2 arguments from stack
```

### 7.2.2 Intrinsics

*Intrinsics* are the internal primitive procedures of the xvm. All intrinsics store the result (if they have one) in the \$ar register. An intrinsic can be used like this:

```
1 ; Allocate memory for 10 bytes.
2 ; Pointer will be in $ar register.
3 PUSH 10
4 INSC AllocMem
```