XièXie Programming Guide

A beginner guidance for the XièXie programming language

Lukas Hermanns

Updated on April 25, 2015

About the Author

My name is Lukas Hermanns (age-group 1990) and I started this project during my studies in 2014. By now I have over 12 years of experience in computer programming, started at the age of 12. I have been writing programs in Basic languages such as QBasic, PureBasic, and Blitz3D; in high level languages such as C, C++, C#, Objective-C, and Java; but also in scripting languages such as JavaScript and Python. I'm actually a preferred programmer in C++ (meanwhile C++11), low level stuff, and graphics programming with shading languages such as GLSL and HLSL.

The XièXie programming language is intended to be simple and not tuned for performance. It was originally designed to be used for scripting in video games, but can also be used for general purposes.

If you like, you can follow me on Twitter, YouTube, GitHub, or Bitbucket.

Contents

I	Th	e XièXie Programming Language	5
1	Intr 1.1 1.2 1.3	oduction What is XièXie? Why is it called "XièXie"? Motivation	6 6 7
	1.5	Motivation	,
2	Syn	tax	8
	2.1	Basics	8
		2.1.1 Commentaries	8
		2.1.2 Identifiers	8
		2.1.3 Literals	9
		2.1.4 String Interpolation	10
	2.2	Operators	10
	2.3	Type Denoters	12
		2.3.1 Built-in Types	12
		2.3.2 Objects	12
		2.3.3 Arrays	12
		2.3.4 Automatic Type Deduction	12
	2.4	Statements	13
		2.4.1 Branch Statements	14
		2.4.2 Loop Statements	15
	2.5	Expressions	17
		2.5.1 Macros	17
3	Clas	sses	18
	3.1	Getting Started	18
	3.2	Declaration Rules for Classes	20
	3.3	Procedures	21
		3.3.1 Procedure Signature	21
		3.3.2 Procedure Overloading	21
		3.3.3 Procedure Overriding	22

			Named Arguments			23
	3.4		outes			24
	3.5		ymous Classes			25
	3.6	Class	Visibilities		•	26
4	Mod	dules				28
	4.1	Using	Modules			28
	4.2		ing Modules			28
II	D	evelop	per Tools			30
5	Con	npiler				31
			nand Line Tool			31
		5.1.1	Output			31
		5.1.2	Script			32
6	Virt	ual Ma				34
	6.1	Execu	ting Programs			34
ш	тт	ow-I c	evel Programming			35
11		OW-LE	ever i rogramming			33
7	Virt		sembler			36
	7.1	Instru	action Set			36
		7.1.1	Registers			36
		7.1.2	OpCodes			37
		7.1.3	Memory Alignment			45
	7.2	Procee	dure Calling			45
		7.2.1	Calling Convention			45
		7.2.2	Intrinsics			46

Part I

The XIÈXIE Programming Language

Chapter 1

Introduction

First of all, "xièxie" is the chinese word for "thanks" or "thank you" (see hantrainerpro.com) and is roughly pronounced "Sh-eh sh-eh".

The design of the XièXie programming language is overall influenced by JAVA, C++, SWIFT, and PYTHON.

1.1 What is XIÈXIE?

The XièXie programming language is a high-level, object-oriented, scripting language with compiler and virtual machine. The XièXie compiler (XXC) translates the XièXie code (XX) to a virtual assembly (XASM), then assembles it to XièXie byte code (XBC), which can then be interpreted by the XièXie virtual machine (XVM).

1.2 Why is it called "XıèXıe"?

Some years, before I started with the development of this compiler, I already had some ideas about a name for it — at this time the compiler was intended to be a cross compiler, i.e. to compile the XièXie code to C++. The first idea I had in mind, was to call it "C+=2" or "C++++" because it should be a more comfortable and easier C++ variant. But this name looked and sounded strange. Another idea was to call it "C power of C" (in mathematical notation C^{C}). But that still was not what I was looking for. Then I remembered me to the Chinese word "Xièxie" which means "Thanks" in English and it sounds a little similar to "CC" which could be seen as a shortcut for C^{C} . That's the XièXie compiler's story about its name.

Or to make a long story short: I saw this word on a napkin in a Chinese restaurant and thought to my self: That's it! :-)

1.3 Motivation

There are many great programming languages out there. Some produce faster code than others, but some are simpler and have a better learning curve. Using a new language doesn't mean to give up a previous one, because every language has its own domain. For example, the performance of interpreted languages such as Python is totally sufficient for many applications. We don't need maximal performance for a script which does some socket connection or text processing for instance. But I would not write the compiler and interpreter for Python in a scripted language. There are also many ways to combine several languages: an interpreter could be used for scripting in video games for instance. But the game engine is written in C++. This is how it is done in Unity3D (see www.unity3d.com). They use Mono as compiler and interpreter framework for C#. But the engine itself is written in C++.

Now \overline{X} is aimed to be used for small scripting purposes, with a gentle learning curve. The great thing about its interpreter is, that it is very tiny and can easily be integrated into existing C or C++ code. This virtual machine only consists of a single code file (written in C99) and can simply be included into any C99 compliant project.

Chapter 2

Syntax

We start out with the syntax.

2.1 Basics

2.1.1 Commentaries

Commentaries are a fundamental part of programming languages and they are nearly identical to those in Java:

```
// Single-line comment
/* Single-line comment */
/*
Multi-line comment
*/
```

Although they are very similar to the commentaries in Java, nested multi-line comments are allowed as well:

```
/*
Outer comment
/* Nested comment */
*/
```

2.1.2 Identifiers

Identifiers (for variables, classes, etc.) must only contain alpha-numeric characters and the underscore. They must also begin with a letter or an underscore. But there is a small exception: they must not begin with __xx___,

because all identifiers with this prefix are reserved for internal use of the compiler.

Valid identifiers are:

- _name_
- FooBar
- number_of_wheels
- Customer01
- ...

Invalid identifiers are:

- naïve
- Foo.Bar
- number-of-wheels
- 3over2
- . . .

2.1.3 Literals

These are the kinds of literals:

- Boolean Literal: true, false
- Integer Literal (Binary): e.g. 0b11001, 0b0000
- Integer Literal (Octal): e.g. 0o24, 0o01234567
- Integer Literal (Decimal): e.g. 3, 12, 999, 1234567890
- Integer Literal (Hexa-Decimal): e.g. 0xff, 0x00, 0xaB29
- Float Literal: e.g. 0.0, 3.5, 12.482
- String Literal: e.g. "Foo Bar", "Hello, World", "\n\t", "1st fragment" "2nd fragment
- Verbatim String Literal: e.g. @"\home\test", @"a ""b"" c"

Integer and float literals may optionally contain the single quotation mark as digit separator, for better readability. If it's used, all separators must satisfy the following rules:

1. For decimal literals, the separators must be **three steps apart** from each other, beginning at the dot.

Example: 12'345, 3.141'592'654, or -27'836.283'74.

2. For non-decimal literals, the separators must be **four steps apart** from each other, beginning at the dot.

Example: 0xff0'214b, 0b100'1101'1011, 0o1234'5670

- 3. A separator must not appear at the beginning or the end of the literal. Counterexample: '123'456'.
- 4. No valid separator must be omitted. Counterexample: 1234'567'890.

2.1.4 String Interpolation

The syntax for *string interpolation* is equal to that in Swift and allows you to easily concatenate strings with other objects or primitive elements, such as integers or floating-point values. If you want to concatenate strings for yourself, it may look like this:

```
var x := 5, y := -3
var s1 := "x = ".append(x).append(", y = ").append(y)
```

This will write "x = 5, y = -3" into s1. The same result can be achieved with the following string interpolation:

```
| var s2 := "x = \(x), y = \(y)"
```

With the escape character '\', everthing in the parenthesis will be read an arbitrary expression, which is appended (or concatenated) to the string on its left. Thus, the string literal s2 is equivalent to s1.

2.2 Operators

The most operators as in Java or C++ are also available in XièXie:

```
/* Arithmetic Operators */
a + b // Addition
a - b // Substration
a * b // Multiplication
a / b // Division
a % b // Modulo
```

```
a << b // Left Shift
a >> b // Right Shift
- a
       // Negate
/* Bitwise Operators */
a & b // Bitwise AND
a | b // Bitwise OR
a ^ b // Bitwise XOR
      // Bitwise NOT
/* Boolean Operators */
not a // Logic NOT
a and b // Logic AND
a or b // Logic OR
/* Relation Operators */
a = b // Equality
a != b // Inequality
a < b // Less
a <= b // Less Or Equal
a > b // Greater
a >= b // Greater Or Equal
```

As the interested reader may have noticed, the *equality operator* is different to that in the most languages. This is because the *copy assignment* operators is := and not =.

In the above example a := b := 3 is invalid, because on the right hand side of the first := must be an expression. But per definition in $Xi \stackrel{\circ}{=} XIE b := 3$ is not an expression, but a statement! The variable list assignment $(a, b, c := \ldots)$ is a comfort functionality, which is only supported for the copy assignment.

The modify-assign operators are available as well:

```
a += b
a -= b
a *= b
a *= b
a /= b
a %= b
a %= b
a <<= b
a >>= b
a &= b
a &= b
a |= b
a ^= b
```

However, they are also not allowed inside another expression.

2.3 Type Denoters

2.3.1 Built-in Types

There are only the following three built-in data types:

```
bool // Boolean type; can be 'true' or 'false'
int // 32-bit signed integral type
float // 32-bit floating-point type
```

2.3.2 Objects

Types for class objects (more about classes in chapter 3) are written as follows:

```
// Empty string
String s

// List of strings
String s1 := "Hello, World", s2 := "Foo", s3 := "Bar"
```

2.3.3 Arrays

The only generic way for lists are the built-in arrays:

```
// Declare array objects with initializer lists
int[] intArray := { 1, 2, 3 }
float[][] floatArrayArray := { { 0.0, 1.5 }, { 3.5, 1.23 } }
String[] stringArray := { "a", "b", "c" }

// Access array elements
String s1 := stringArray[0]
String s2 := stringArray[intArray[0]]
float[] floatArray := floatArrayArray[1]
```

2.3.4 Automatic Type Deduction

Whereas automatic type deduction in C++11 is a very extensive language feature, in XièXie it can be summarized in this section. There are two keywords for automatic type deduction: var and const. As the name implies var denotes a variable type and const denotes a constant type. The latter type is the only way to define constants in XièXie. Here are a few examples:

2.4 Statements

Unlike C++ and Java, there are no semicolons in XièXie to terminate statements. Only the regular for-statement has two semicolons, to separate the initializer statement, the condition expression, and the increment statement. This means the compiler (or rather the *parser*, which reads the source code) always knows when a statement or expression is complete. But it also means that you — the programmer — can do weird things with this syntax. Consider the following code sample:

```
int a:=3
-4
,b:=-
5+2 int c
```

This is valid XIÈXIE code and it contains only two statements! If we write it in a more common convention, it may look like this:

```
int a := 3-4,
    b := -5+2
int c
```

Hence, the readability of your code is up to you and your programming style :-). The only language I've worked with, which forces you to practice better readability is PYTHON. Actually a great principle, but with XIÈXIE you have complete freedom.

The absence of statement terminators is the reason for the *double paren* syntax of attributes:

```
[[attribute]]
```

To understand why this is the case, let's assume attributes are written with a single paren and consider the following class declaration:

```
class Widget {
   const c := 0 // Initialize member constant 'c' with 0
   int v := c // Initialize member variable 'v' with constant 'c'
   [final] // Mark next procedure with attribute 'final'
   void proc() {}
```

}

Now this doesn't seem very complex. But the parser runs into trouble when reading [final]. This is because the parser reads it as follows:

```
int v := c[final]
```

But c is not an array. This is why attributes are written with a double paren, because array accesses never begin with '[['. They may end with ']]', but this is not important for the parser.

In the following sections, we will see several types of statements, which are:

- *Branch* Statements: **if**, **switch**.
- Loop Statements: for, foreach, while, do-while, repeat.
- Control Transfer Statements: break, continue, return.

2.4.1 Branch Statements

A branch statement represents a branch in the control flow of a program, i.e. the program may pass through different paths during execution, depending on which branch is taken.

if Branch

```
// If x is greater than y, then execute the following code block
if x > y  {
    doSomethingUseful()
}
// Nested boolean expression: x must be equal to y,
// and the condition inside the parentheses must be 'false'
if x = y and not (x != 3 \text{ or } y = -7) {
    thenDoThis()
} else {
    otherwiseDoThat()
// Classic if/elseif/else statement
if conditionA {
   /* ... */
} else if conditionB {
    /* ... */
} else if conditionC {
    /* ... */
} else {
    /* ... */
```

switch Branch

```
// Switch statement with many cases (only integers are allowed)
int idx := getSwitchIndex()
switch idx {
    case 1:
        print("case 1") // idx = 1
        print("case 2") // idx = 2
    case 3 .. 10:
        print("case 3") // idx >= 3 and idx <= 10
    case 11, 20 .. 25, 30:
        print("case 4") // idx = 11 or (idx >= 20 and
                        // idx <= 25) or idx = 30
    default:
        print("default case")
}
// Switch with break statements
const caseIndex := 11
int idx := getSwitchIndex()
switch idx {
    case 1:
        int x := 42 // each case has its own scope
    case 2 .. 10:
        int x := idx*2 // declares a new variable in this scope
        if x > 10 {
            break // jump out of switch statement
        print("hi there!")
    case caseIndex:
        print("case \(caseIndex)")
}
```

2.4.2 Loop Statements

It follows several examples of loop statements.

for Loop

```
// This regular for loop iterates 'i' from 0 to 9 (similar to Java)
for int i := 0 ; i < 10 ; i++ {
    // Infinite loop (also similar to Java)
    for ;; {
        if i >= 0 {
            // Break infinite loop
            break
        }
    }
```

```
// Inner iteration variable 'j' is implicit initialized to 0.0
float j
for ; j < 3.5 ; {
        j += 0.5
    }
}</pre>
```

Range-Based for Loop

foreach Loop

```
// Iterate over array with elements 1, 2, and 3
foreach i : { 1, 2, 3 } {
    print(i)
}

// Iterate over a 'superList' from type (array of arrays of strings)
String[][] superList := { "a", "b" }, { "c" } }
foreach list : superList {
    // Iterate over all elements in the current sub list
    foreach str : list {
        // Do something with this string
        print(str)
    }
}
```

repeat Loop

```
// Repeat for unconditional iterations
// (This is internally a "ForEver"-loop)
repeat {
    // Condition to break the loop
    if magicFunction() {
        // Break loop
        break
```

```
}
}
// Do something 10 times (with invisible index variable)
// -> This is internally a 'range-based for loop',
// i.e. equivalent to 'for i : 1 .. 10'
repeat 10 {
    doSomething()
}
```

while Loop

```
// Regular while loop
while magicFunction() {
   doSomething()
}
```

do/while Loop

```
// Regular do-while loop
do {
   doSomething()
} while magicFunction()
```

2.5 Expressions

2.5.1 **Macros**

XIÈXIE does not support the declaration of macros. However, there are a few built-in macros, which represent additional reserved identifiers:

- __FILE__ String which contains the current filename.
- __CLASS__ String which contains the current class name.
- __PROC__ String which contains the current procedure name.
- __LINE__ Integer which contains the current line number.
- __DATE__ String which contains the current date and time.
- __VERSION__ Integer which contains the compiler version number (e.g. 200 for version "2.00").

Chapter 3

Classes

A XièXie program can only consist of imports, modules, and class declarations. And classes can only be defined in the global scope. That means every procedure must be defined inside a class and inner classes are currently not supported.

3.1 Getting Started

To get started, take a look at the following example program which prints the classical phrase "Hello, World!" onto the standard output:

```
// XieXie Hello World Program
import System
class HelloWorld {
    static void main() {
        System.out.writeLine("Hello, World!")
    }
}
```

This merely writes the line "Hello, World!" to the standard output. Let's take a closer look at each line.

Line 2 imports the "System.xx" file from the XièXie standard library:

```
import System // either this ...
import "System.xx" // ... or this
```

If the imported file is in another directory, the string version of import is the only choice. This can also be omitted if the file is added to the compilation process. The files for the classes Object, String, Array, and Intrinsics are always implicit imported, because they are 'internal' classes the compiler knows generally. Note that verbatim strings are allowed wherever string literals are allowed, i.e. the following example is valid XièXie code:

```
import "C:\\Program Files\\Test1.xx" // either this ...
import @"C:\Program Files\Test1.xx" // ... or this
```

The import keyword is different to that in Java and also different to the #include directive in C++. Although it takes a filename as parameter (like C++'s #include), it does not *include* the file in place. Whenever an import is read by the *parser*, the filename is added to the set of import files. After all source files have been read, which were passed as input to the compiler, all import files will be read next. This will be repeated until no new files are added to the set. Consider this is a *set* of files, i.e. several import commands may occur with the same filename, but it will be read only once. This is why the above sample is valid XièXie code. It also means that recursive imports are allowed as well:

```
// File1.xx
import "File2.xx"
```

```
// File2.xx
import "File1.xx"
```

Line 3 declares the class HelloWorld which implicit inherits from the base class Object, like it is done in Java:

```
3 class HelloWorld { /* ... */ }
```

To inherit from other classes, just write a colon and the identifier of the base class:

```
class SubClass : BaseClass { /* ... */ }
```

There is no multiple inheritance like in C++ or interfaces like in Java! The next line declares the procedure main.

```
4 | static void main() {
```

This is the main entry point for the program. There can only be one main entry point, but there are several possible signatures for this procedure:

```
// No return value, no arguments
static void main() { /* ... */ }

// No return value, arguments
static void main(String[] args) { /* ... */ }

// Return value, no arguments
static int main() { /* ... */ }

// Return value, arguments
static int main(String[] args) { /* ... */ }
```

The last line of code prints the message to the standard output:

```
5 System.out.writeLine("Hello, World!")
```

System is a class from the standard XièXie library, out is a *static* member from the type 'OutputStream', and writeLine is a function which takes a string as input.

3.2 Declaration Rules for Classes

In XièXie there is no need for *forward declarations*. Everything can be declared in the respective scope and is accessible throughout the entire program (except private scope). This is why the following code is valid:

```
// First declare sub class
class SubClass : BaseClass { /* ... */ }

// Then declare base class
class BaseClass { /* ... */ }
```

The same applies for procedure declarations:

```
class B {
    static void procB1() {
        A.procA() // no forward declaration required ...
        B.procB2() // ... same here
    }
    static void procB2() { /* ... */ }
}
class A {
    static void procA() { /* ... */ }
}
```

This works because the *context analyzer* of the compiler works in several phases:

- 1. Class symbols are registered in global scope.
- 2. Class signatures are analyzed (attributes and base class).
- 3. Class inheritance is verified (check for cycles).
- 4. Member procedures are registered in respective class scope.
- 5. Member variables are registered in respective class scope.
- 6. Return type of all member procedures is analyzed.
- 7. Run-time type information (RTTI) for the entire class hierarchy is generated.
- 8. Procedure code is analyzed.

3.3 Procedures

In XièXie we talk about *procedures* (somethings called *methods*), because in the strict sense *functions* have no side effects. Functions have only input parameters which are calculated to a result. But in XièXie every procedure can have side effects, meaning that they can modify the program state (with static variables for instance).

3.3.1 Procedure Signature

A *procedure signature* is the extended identification of a procedure beyond its identifier string. The signature consists of the identifier string, its parameter list, and the procedure's return type. However, the return type is never used for identification.

3.3.2 Procedure Overloading

XièXie supports overloading of procedures. This means the same identifier can be used several times for procedures inside a class declaration (including its inheritance hierarchy). This requires that the following rules are satisfied:

All procedures with the same identifier inside a class declaration can be distinguished by their parameter count or parameter types.

Here is an example of procedure overloading.

When overloading procedures, try to avoid ambiguities with default arguments. Adding default arguments to all parameters of all overloaded procedures is allowed, but procedure calls may be ambiguous for the compiler:

3.3.3 Procedure Overriding

XièXie supports overriding of procedures. This means the same procedure signature can be used inside a class and its base class. The procedure calls of overloaded and overridden procedures require that the following rules are satisfied:

If a procedure with identifier *P* is declared inside a class *C*, a procedure with the same identifier is declared in its base class *B* but with another signature, and another procedure inside class *C* calls the procedure *P* from class *B*, then the identifier super must be specified in front of the call.

To better understand this awkward definition, take a look at this example:

```
class B {
    int f(int x, int y) {
        return 1
    int g() {
        return 2
    }
class S : B {
    int f(int x) {
        return 3
    void caller() {
        int a := f(0)
                              /* Equivalent to 'this.f(0)' */
        int b := super.f(0, 0) /* 'super' is required,
                                  due to overloaded procedure 'f' */
                               /* No need for 'super',
        int c := g()
                                  because 'g' is not overloaded */
    }
}
```

3.3.4 Named Arguments

Named arguments, like in C#, are supported as well. They also have the same rules as in C#, i.e. if you started to write named arguments in a procedure call, you need to finish it till the last argument. Here is an example and counter-example:

```
// Example:
class Widget {
    void f(int x := 1, int y := 2, int z := 3)
    void call() {
        int y := 12
        f()
                            // Use all default arguments
        f(y: -5)
                            // Equivalent to f(1, -5, 3)
        f(y: y)
                            // Equivalent to f(1, y, 3)
                           // Equivalent to f(-1, 2, 0)
        f(-1, z: 0)
        f(z: 1, y: 2, x: 3) // Equivalent to <math>f(3, 2, 1)
    }
}
// Counter-example:
class Widget {
    void f(int x := 1, int y := 2, int z := 3)
    void call() {
        f(1, x: 1) /* Error, parameter 'x' already assigned */
        f(1, y: 2, 3) /* Error, can not start with named
                         arguments and continue with unnamed */
    }
}
```

But what are named arguments good for? Consider the following function signature:

```
1
    // User account class.
2
    class UserAccount {
3
        init(
            String name := "<unnamed>",
4
5
            int age := 0,
            String location := "",
6
7
            String city := "",
8
            int postalCode := 0
9
            int flags := 0
10
        /* · · · */
11
12
   | }
13
14
    // Creates an account, only initialize with basic information.
15
    class AccountManager() {
16
        // Alternative 1
17
        UserAccount createAccountAlt1(String name, int flags := 0) {
18
            return new UserAccount(name: name, flags: flags)
19
        }
20
```

```
21  // Alternative 2
22  UserAccount createAccountAlt2(String name, int flags := 0) {
23     return new UserAccount(name, 0, "", "", 0, flags)
24  }
25  }
```

In this example alternative 2 (Line 23) is not much more to type. But alternative 1 (Line 18) is much better to read :-). You see immediately which arguments are passed to which parameters. No matter if the parameter order will change, your procedure call will still work as suggested. Moreover, the default arguments in the procedure declaration may change, but your procedure call will still use them correctly.

3.4 Attributes

There are several attributes for class-, procedure-, and variable declarations:

```
// Class A is marked as 'deprecated'.
[[deprecated]]
class A {
    // Declare some procedure
    void p() {}
}
/* Class B is marked as 'deprecated' with a hint.
   Since B is deprecated as well, the deprecation
   of A is ignored here. */
[[deprecated("hint...")]]
class B {
    /* Mark 'p' to override 'A.p'. If 'p' does not match the
       signature of procedure 'A.p', the compiler will
       throw an error message. */
    [[override]]
    void p() {}
}
/* Class C is 'final', i.e. no class can inherit from C.
   Because A is deprecated but not C,
   the compiler will throw a warning. */
[[final]]
class C : A {}
```

Here is a summary of all attributes:

- deprecated, deprecated(String hint) Marks a class, procedure, or member-variable as deprecated.
- **export**, **export(String label)** Exports the address of a *static procedure*. The default label is "CLASS.PROCEDURE", where CLASS is the class

name, and PROCEDURE is the procedure name (without name mangling). This can be used to start the program from a specific static procedure. This should be some kind of an entry point!

- bind, bind(String name) Binds a module of the specified name. The
 default name is: "NAME/NAME", where NAME is the name of the module.
 Binding a module will make it loaded automatically when the program
 is started within the xvm.
- **override** Overrides a procedure from a class in its inheritance hierarchy.
- **final** Makes a class final, which disables to inherit from this class.

3.5 Anonymous Classes

XièXie supports the declaration of anonymous classes. This can be quite comfortable when a class needs to be instantiated only once:

```
class Hello {
    /* No code block after procedure declaration inside
       a non-extern class implies an abstract procedure */
    String say()
}
class EnglishHello : Hello {
    [[override]]
    String say() { return "Hello" }
}
class Main {
    static void main() {
        /* Allocate class "EnglishHello" */
        var englishHello = new EnglishHello()
        /* Allocate anonymous class
          (internal name "__xx__AnonymousClass0") */
        var germanHello = new Hello() {
            [[override]]
            String say() { return "Hallo" }
        }
    }
}
```

Now it should be clear why identifiers with the prefix "__xx__" are reserved.

3.6 Class Visibilities

Like in C++ and Java, XièXie provides visibilites for all members within a class. The syntax for these visibilites are a mixture of C++ and Java. You can declare the visibility for each class member like in Java and/or for a whole bunch of members like in C++:

```
class C {
public:
               int
                       publicVar0
               float
                       publicVar1
    private
               int
                       privateVar0
               String
                       publicVar2
    protected int
                       protectedVar0
    private
               float[] privateVar1
               int[]
                       publicVar3
private:
               int
                       privateVar2
    public
              bool
                       publicVar4
               bool[]
                       privateVar3
}
```

Unlike in common object-oriented languages, the default visibility in XièXie is public. This is because there are only classes and no structures like in C++. So a small class, which only has variables and acts like a data structure can be written with as less effort as possible:

```
// Examples of simple data structures
class Vector3 { float x, y, z }
class FileHeader {
   int magicNumber
   int formatVersion
}
```

Like in C++, there are also 'friends'. And like in C++ as well, friendship is not inherited:

Chapter 4

Modules

A *module* consists of a shared library (*.dll on Windows and *.so on GNU/Linux) and a XièXie module file. They are similar to Python modules.

4.1 Using Modules

Module declarations are similar to external classes. But they can only have static procedures. Here is an example:

```
module MyModule {
    static void doSomething(int x)
}
```

Using this in your XIÈXIE code will then look like this:

```
import MyModule
class MyClass {
    static void main() {
        MyModule.doSomething(42)
    }
}
```

4.2 Creating Modules

Modules must be written in plain C or C++. A minimal module would consist of three files. Supposed our example module is named "MyModule" we would have these files:

• MyModule.c Module code written in C.

- MyModule.dll (Windows)/ *.so (GNU/Linux) Shared library which runs the module code.
- MyModule.xx XièXie code file which declares the module interface.

The C code file must implement the following function interfaces to be a valid XièXie module:

```
// Returns the number of module procedures.
int xx_module_proc_count();

// Returns the procedure for the respective index.
XVM_INVOCATION_PROC xx_module_fetch_proc(int index);

// Returns the procedure identifier for the respective index.
const char* xx_module_fetch_ident(int index);
```

Our simple module example from above could be implemented as follows:

```
// Include XieXie module header
#include <xiexie/xx module.h>
// This is our module procedure "soSomething",
// all module procedures must have this interface
void doSomething(XVM_Env env) {
    // Get 1st parameter from XVM environment
    int x = XVM_ParamInt(env, 1);
    // Do something with 'x'
    printf("input parameter 'x' is %i\n", i);
    // Pop arguments from stack (1 for the single parameter 'x')
    XVM_ReturnVoid(env, 1);
}
// Implement the module interface
static XVM_INVOCATION_PROC procList[] = {
    { "doSomething", doSomething },
};
XVM_IMPLEMENT_MODULE_INTERFACE(procList);
```

Part II Developer Tools

Chapter 5

Compiler

5.1 Command Line Tool

The compiler can be used as command line tool. The appropriate program is named xxc. Enter xxc help in a command line to see the manual pages.

In contrast to most other command line tools, the commands for xxc don't have the '-' prefix. Instead the command options use this prefix. Here is an example:

```
xxc C -f FILE1 -f FILE2 -0
```

The above command line uses the C command (also compile) with the flags '-f' and '-O'.

5.1.1 Output

The compiler tries to show you where an error or warning occurred. If possible, it is highlighted with a *line marker*. Consider the following code sample:

```
class Foo {
    void Bar() {
        int x := y
    }
}
```

Now consider the compiler processes this code with the following command line:

```
xxc C -f Core.xx -f Foo.xx
```


```
parse file "Core.xx" ...
parse file ".(CompilerOutputTest1.xx" ...
context analysis ...
2 error(s), 0 warning(s):

context arror (3:11 - 3:12) -- undeclared identifier "y"

>> unt x != y

>> int x != y
```

(a) Output in PowerShell on Windows 7



(b) Output in Terminal on Kubuntu GNU/Linux

Figure 5.1: Compiler output for the above code sample.

Then the compiler indicates that the variable 'y' was not declared (see Figure 5.1). As you can see, some errors may produce multiple outputs. In this case, 'y' is undeclared which produces two error outputs here:

- 1. The identifier 'y' is undeclared, so the appearance in the initializer expression is invalid.
- 2. The type of the initializer expression for the declaration of 'x' can not be deduced, so the type check fails.

5.1.2 Script

If xxc has no arguments, it will switch to the *immediate script* mode. This allows the user to immediately script XièXie code inside the command line (see Figure 5.2).

```
Windows PowerShell

PS C:\Users\Lukas Hermanns> xxc
XieXie Compiler 2.0 Alpha
Copyright (c) 2014-2015 Lukas Hermanns
01 for i : { 42, 7, -3, 9 } {
02    print(i)
03 }_
```

Figure 5.2: Compiler immediate script mode.

Chapter 6

Virtual Machine

The XièXie Virtual Machine (XVM) is a separated program (named xvm), written in pure C99.

6.1 Executing Programs

To execute (or run) a virtual program, just enter the filename of an *.xbc file into the xvm:

xvm HelloWorld.xbc

Part III Low-Level Programming

Chapter 7

Virtual Assembler

The XièXie Virtual Assembler (XASM) is the low level language which is used as interface to the xvm. It's a very low level language, with similarities to ARM® Assembler.

7.1 Instruction Set

7.1.1 Registers

The xvm is a register machine, i.e. it uses registers as operands for its instructions instead of the stack. This commonly increases performance but is a little more tricky to use, when you run out of registers. Each register index inside an instruction is stored with 4 bits, thus there are 16 registers and this is the list:

- 1. **\$r0** General Purpose Register 0.
 - :
- 25. \$r24 General Purpose Register 24.
- 26. **\$ar** Argument Return.
- 27. **\$xr** Extended Register.
- 28. \$gp Global Pointer.
- 29. **\$cf** Conditional Flag.
- 30. **\$1b** Local Base Pointer.
- 31. \$sp Stack Pointer.
- 32. **\$pc** Program Counter.

All pointer registers and the program counter (i.e. \$gp, \$1b, \$sp, and \$pc) are all 'real' pointers to the memory in your operating system.

7.1.2 OpCodes

Each instruction stores its mnemonic (Greek 'memory' $\hat{}$ instruction identifier) in the first 6 bits. This provides a maximum of 64 instructions. However, only 52 mnemonics are currently in use.

	, + t s	Describution	Bitwise AND (D := L & R).	Bitwise OR (D := L R).	Bitwise XOR (D := L $^{\circ}$ R).	Integer add. (D := L + R).	Integer sub. (D := L - R).	Integer mul. (D := $L \times R$).	Integer div. (D := L \div R).	Modulo $(D := L \% R)$.	Shift left (D := L \times R).	Shift right (D := $L \times R$).	Float add. (D := L + R).	Float sub. (D := L - R).	Float mul. (D := $L \times R$).	Float div. (D := L \div R).
3-Register Instruction OpCodes (00)	Unused	100	000000000000	00000000000	000000000000	00000000000	00000000000	00000000000	00000000000	00000000000	000000000000	00000000000	00000000000	000000000000	00000000000	00000000000
truction OpC	RSource	1511	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR
Register Ins	LSource	2016	TTTTT	TTTTT	TTTTT	TTTTT	TTTTT	TTTTT	TTTTT	TTTTT	TTTTT	TTTTT	TTTTT	TTTTT	TTTTT	TTTTT
3-	Dest.	2521	DDDDD	ррррр	рррррр	ррррр	ррррр	ррррр	ррррр	рррррр	рррррр	рррррр	ррррр	рррррр	рррррр	DDDDD
	OpCode	3126	000001	000010	000011	000100	000101	000110	000111	001000	001001	0 0 1 0 1 0	001011	0 0 1 1 0 0	001101	0 0 1 1 1 0
	M	THEMOTIT	AND	OR	XOR	ADD	SUB	MUL	DIV	MOD	STT	SLR	ADDF	SUBF	MULF	DIVF

		-2	-Register Ins	2-Register Instruction OpCodes (01)	
M	0pCode	Dest.	Source	Unused or Value	2000
rinemonic	3126	2521	2016	150	Describution
MOV	0 1 0 0 0 0	рррррр	SSSSS	000000000000000000	Move $(D := S)$.
NOT	0 1 0 0 0 1	ррррр	SSSSS	00000000000000000	Bitwise NOT (D := \sim S).
FTI	0 1 0 0 1 0	пррррр	SSSSS	000000000000000000	Float to integer (D := (int)S).
ITF	0 1 0 0 1 1	пррррр	SSSSS	000000000000000000	Integer to float $(D := (float)S)$.
AND	0 1 0 1 0 0	пппппп	SSSSS		Bitwise AND $(D := S & V)$.
OR	0 1 0 1 0 1	пррррр	SSSSS	$ \land \land$	Bitwise OR (D := $S \mid V$).
XOR	0 1 0 1 1 0	пррррр	SSSSS	$ \land \land$	Bitwise XOR (D := $S^{\circ}V$).
ADD	0 1 0 1 1 1	пррррр	SSSSS	\(\lambda \) \(Modulo (D := 5% V).
SUB	0 1 1 0 0 0	прпррр	SSSSS		Shift left (D := $S \times V$).
MUL	0 1 1 0 0 1	пррррр	SSSSS	$ \land \land$	Shift right (D := $S \gg V$).
DIV	0 1 1 0 1 0	пррррр	SSSSS	\(\lambda \) \(Integer add. $(D := S + V)$.
MOD	0 1 1 0 1 1	00000	SSSSS	\ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \	Integer sub. (D := $S - V$).
STT	0 1 1 1 0 0	пппппп	SSSSS	\ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \	Integer mul. (D := $S \times V$).
SLR	0 1 1 1 0 1	ααααα	SSSSS	\ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \	Integer div. (D := $S \div V$).
CMP	0 1 1 1 1 0	X X X X X	YYYYY		Integer comparison ($\rightarrow $ \$cf).
CMPF	0 1 1 1 1 1	XXXXX	YYYY	000000000000000000	Float comparison ($\rightarrow $ \$cf).

		1	1-Register Instruction OpCodes (100)	
W	OpCode	Reg.	Unused, Value, or Offset	30000
rinelloni	3126	2521	3126 2521 20	Descributoii
PUSH	100000	RRRRR	1 0 0 0 0 0 0 R R R R R 0 0 0 0 0 0 0	Push register onto stack.
POP	100001	RRRRR	100001 RRRR 000000000000000000000000000	Pop register from stack.
INC	100010	RRRRR	$1\ 0\ 0\ 0\ 1\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\$	Increment integer (D++).
DEC	100011	RRRRR	$1\ 0\ 0\ 0\ 1\ 1\ R\ R\ R\ R\ R\ R\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\ 0\$	Decrement integer (D–).
MOV	100100	RRRRR	$1 \ 0 \ 0 \ 1 \ 0 \ 0 \\$	Move integer (D := V).
LDA	100101	RRRRR	100101 RRRRR 00000000000000000000000000	Load address from program
				10722000

_	_				_		_	_		_		
	Doctoriation	Desci ipcion	Jump.	Jump if equal.	Jump if not-equal.	Jump if greater.	Jump if less.	Jump if greater or equal.	Jump if less or equal.	Push dynamic link (\$1b and \$pc)	onto stack; Set \$1b to new stack	frame; Jump to address.
Jump Instruction OpCodes (101)	Offset	3126 2521 20	101000 RRRR 000000000000000000000000000	101000 RRRR 000000000000000000000000000	101010 RRRR 000000000000000000000000	101011 RRRR 000000000000000000000000000	101100 RRRRR 00000000000000000000000	1 0 1 1 0 1 R R R R R 0 0 0 0 0 0 0 0 0 0 0 0 0	1 0 1 1 1 0 R R R R R 00000000000000		1 0 1 1 1 1 R R R R R 0 0 0 0 0 0 0 0 0 0 0 0 0	
	Reg.	2521	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR	RRRRR		RRRRR	
	OpCode	3126	101000	101001	101010	101011	101100	101101	101110		101111	
	Mr. romorn	יווופוווסוודר	JMP	JE	JNE	JG	JL	JGE	JLE		CALL	

		Los	ad/Store Inst	Load/Store Instruction OpCodes (1100)	
M	OpCode	Reg.	Addr.	Offset	70000
THEMOTIT	3126	2521	2016	150	הפכן זף רוסוו
LDB	110000	RRRRR	AAAAA	1 1 0 0 0 0 R R R R R A A A A A A A A A A A	Load byte from memory.
STB	110001	RRRRR	AAAAA	110001 RRRR AAAA 0000000000000000	Store byte to memory.
LDW	110010	RRRRR	AAAAA	110010 RRRR AAAA 0000000000000000	Load word from memory.
STW	110011	RRRRR	AAAAA	1 1 0 0 1 1 RRRR AAAA 0 0 0 0 0 0 0 0 0 0 0 0 0 0 Store word to memory.	Store word to memory.

Special-1 Instruction OpCode Unused or Value State Sta							
		Doccerintion	תפסרו ולה רוסוו	Stop program execution.	Push integer onto stack.	Invoke external procedure.	Call intrinsic.
	Special-1 Instruction OpCodes	Unused or Value	250				\ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \ \
		OpCode	3126	0000000	1 1 1 0 0 0	1 1 1 0 0 1	1 1 1 0 1 0
		II	THEMIONE		PUSH	INVK	INSC

1			I.	٩	p	<u></u> 9	
	Description		Pop R wrods from stack and buffer	them; Pop current stack frame; Pop	A words from stack; Push stored	R words back onto stack; Restore	dynamic link (\$1h and \$nc)
, , , , , , , , , , , , , , , , , , ,	Argument Size	2				A A A A A A A A A A A A A A A A A A A	
	DpCode Result Size 16	27				KKKKKKKKK	
	OpCode	21					
	Mnemonic				БЕТ	RE I	

7.1.3 Memory Alignment

Some instructions use byte alignment and some others use word alignment.

7.2 Procedure Calling

7.2.1 Calling Convention

In XASM, the calling convention provides that the *callee* removes the procedure arguments from the stack. After any procedure call, the *dynamic link* is stored at the beginning of the *stack frame* and reserves the first 8 bytes.

dynamic link

Consists of two words: 32-bit value of the \$1b register *before* the procedure call and 32-bit value of the \$pc register *before* the procedure call.

Local variables can be store by increasing the *stack pointer* (\$sp) and load/store instructions at the *local base pointer* (\$1b) plus 8 bytes (after the *dynamic link*). Here is an example:

```
; void proc() { ... }
1
2
   proc:
3
     ADD $sp, $sp, 8 ; Allocate memory on the stack ...
4
                        ; ... for two words (2*4 bytes)
     MOV $r0, 42 ; Store value 42 in register $r0
5
     STW $r0, ($lb) 8 ; Store register $r0 in local scope ...
6
7
                         ; ... at first position (after dynamic link)
     ADD $r0, $r0, 10 ; Increment value of register $r0
8
9
     STW $r0, ($1b) 12; Store register $r0 in local scope ...
10
                          ; ... at second position
11
```

The arguments for a procedure call are meant to be pushed onto the stack from *right-to-left*. In this way the arguments can be accessed from *left-to-right*. Here is an example:

```
1  ; int proc(int x, int y) { return x+y }
2  proc:
3  LDW $r0, ($lb) -4 ; Fetch first argument (x) from stack
4  LDW $r1, ($lb) -8 ; Fetch second argument (y) from stack
5  ADD $ar, $r0, $r1 ; Calculate ar := x + y
6  RET 2  ; Return: $ar and pop 2 arguments from stack
```

7.2.2 Intrinsics

Intrinsics are the internal primitive procedures of the xvm. All intrinsics store the result (if they have one) in the \$ar register. An intrinsic can be used like this:

```
1 ; Allocate memory for 10 bytes.
2 ; Pointer will be in $ar register.
3 PUSH 10
INSC AllocMem
```