XièXie Programming Guide

A beginner guidance for the XièXie programming language

Lukas Hermanns

Updated on March 25, 2015

About the Author

My name is Lukas Hermanns (age-group 1990) and I started this project during my studies in 2014. By now I have over 12 years of experience in computer programming, started at the age of 12. I have been writing programs in Basic languages such as QBasic, PureBasic, and Blitz3D; in high level languages such as C, C++, C#, Objective-C, and Java; but also in scripting languages such as JavaScript and Python. I'm actually a preferred programmer in C++ (meanwhile C++11), low level stuff, and graphics programming with shading languages such as GLSL and HLSL.

The XièXie programming language is intended to be simple and not tuned for performance. It was originally designed to be used for scripting in video games, but can also be used for general purposes.

If you like, you can follow me on Twitter, YouTube, GitHub, or Bitbucket.

ToDo List

The compiler, and partially the virtual machine, are not yet completed. Some rules and explanations in this report may change over time. Here is a rough ToDo-list:

Parser	
Implementation of parser against grammar specification	almost done
Member procedure calls as statements	not yet implemented
Post fix array access	done
Destructor	not yet implemented
Named Parameters	done
Context Analyzer	
Expression Type Check	done
Cast Type Check	incomplete
Automatic Type Deduction	done
Procedure Overloading	done
Procedure Overriding	done
Named Parameters	not yet implemented
Static and Non-Static Procedure Behavior	incomplete
Class and Procedure Attributes	not yet implemented
Code Generator	
Common CFG and TAC Generation from DAST	incomplete
If Statement	almost done
ForEver Statement	done
XASM Backend	
Final code generation for XASM	not yet implemented

Contents

Part I

The XièXie Programming Language

Chapter 1

Introduction

First of all, "xièxie" is the chinese word for "thanks" or "thank you" (see hantrainerpro.com) and is roughly pronounced "Sh-eh sh-eh".

The design of the XièXie programming language is overall influenced by Java, C++, and Python.

1.1 What is XièXie?

The XièXie programming language is a high-level, object-oriented, scripting language with compiler and virtual machine. The XièXie compiler (XXC) translates the XièXie code (XX) to a virtual assembler (XASM), then assembles it to XièXie byte code (XBC), which can then be interpreted by the XièXie virtual machine (XVM).

1.2 Why is it called "XièXie"?

Some years, before I started with the development of this compiler, I already had some ideas about a name for it — at this time the compiler was intended to be a cross compiler, i.e. to compile the XièXie code to C++. The first idea I had in mind, was to call it "C+=2" or "C++++" because it should be a more comfortable and easier C++ variant. But this name looked and sounded strange. Another idea was to call it "C power of C" (in mathematical notation C^C). But that still was not what I was looking for. Then I remembered me to the Chinese word "Xièxie" which means "Thanks" in English and it sounds a little similar to "CC" which could be seen as a shortcut for C^C. That's the XièXie compiler's story about its name.

Or to make a long story short: I saw this word on a napkin in a Chinese restaurant and thought to my self: That's it! :-)

1.3 Motivation

There are many great programming languages out there. Some produce faster code than others, but some are simpler and have a better learning curve. Using a new language doesn't mean to give up a previous one, because every language has its own domain. For example, the performance of interpreted languages such as Python is totally sufficient for many applications. We don't need maximal performance for a script which does some socket connection or text processing for instance. But I would not write the compiler and interpreter for Python in a scripted language. There are also many ways to combine several languages: an interpreter could be used for scripting in video games for instance. But the game engine is written in C++. This is how it is done in Unity3D (see www.unity3d.com). They use Mono as compiler and interpreter framework for C#. But the engine itself is written in C++.

Now XièXie is aimed to be used for small scripting purposes, with a gentle learning curve. The great thing about its interpreter is, that it is very tiny and can easily be integrated into existing C or C++ code. This virtual machine only consists of a single code file (written in C99) and can simply be included into any C99 compliant project.

Chapter 2

Syntax

We start out with the syntax.

2.1 Basics

2.1.1 Commentaries

Commentaries are a fundamental part of programming languages and they are nearly identical to those in Java:

```
1 // Single-line comment
2
3 /* Single-line comment */
4
5 /*
6 Multi-line comment
7 */
```

Although they are very similar to the commentaries in Java, nested multi-line comments are allowed as well:

```
1 /*
2 Outer comment
3 /* Nested comment */
4 */
```

2.1.2 Identifiers

Identifiers (for variables, classes, etc.) must only contain alpha-numeric characters and the underscore. They must also begin with a letter or an underscore. But there is a small exception: they must not begin with `__xx__`, because all identifiers with this prefix are reserved for internal use of the compiler.

Valid identifiers are:

- `_name_`
- `FooBar`
- `number_of_wheels`
- `Customer01`
- ...

Invalid identifiers are:

- `naïve`
- `Foo.Bar`
- `number-of-wheels`
- `3over2`
- ...

2.1.3 Literals

These are the kinds of literals:

- Boolean Literal: `true`, `false`
- Integer Literal (Binary): e.g. `0b11001`, `0b0000`
- Integer Literal (Octal): e.g. `0o24`, `0o01234567`
- Integer Literal (Decimal): e.g. `3`, `12`, `999`, `1234567890`
- Integer Literal (Hexa-Decimal): e.g. `0xff`, `0x00`, `0xaB29`
- Float Literal: e.g. `0.0`, `3.5`, `12.482`
- String Literal: e.g. `"Foo Bar"`, `"Hello, World"`, `"\n\t"`, `"1st fragment"␣"2nd fragment`
- Verbatim String Literal: e.g. `@"\home\test"`, `@"a ""b"" c"`

Integer and float literals may optionally contain the single quotation mark as digit separator, for better readability. If it's used, all separators must satisfy the following rules:

1. For decimal literals, the separators must be **three steps apart** from each other, beginning at the dot.
   Example: `12'345`, `3.141'592'654`, or `-27'836.283'74`.
2. For non-decimal literals, the separators must be **four steps apart** from each other, beginning at the dot.
   Example: `0xff0'214b`, `0b100'1101'1011`, `0o1234'5670`
3. A separator must not appear at the beginning or the end of the literal.
   Counterexample: `'123'456'`.
4. No valid separator must be omitted.
   Counterexample: `1234'567'890`.

2.2 Operators

The most operators as in Java or C++ are also available in XièXie:

```
/* Arithmetic Operators */
a + b   // Addition
a - b   // Substration
a * b   // Multiplication
a / b   // Division
a % b   // Modulo
a << b  // Left Shift
a >> b  // Right Shift
-a      // Negate

/* Bitwise Operators */
a & b   // Bitwise AND
a | b   // Bitwise OR
a ^ b   // Bitwise XOR
~a      // Bitwise NOT

/* Boolean Operators */
not a   // Logic NOT
a and b // Logic AND
a or b  // Logic OR

/* Relation Operators */
a = b   // Equality
a != b  // Inequality
a < b   // Less
a <= b  // Less Or Equal
a > b   // Greater
a >= b  // Greater Or Equal
```

As the interested reader may have noticed, the *equality operator* is different to that in the most languages. This is because the *copy assignment* operators is `:=` and not `=`.

```
x := 5       // ok, set x to 5
a, b, c := 4 // ok, set a, b, and c to 4
a := b := 3  // error, b := 3 is not an expression
```

In the above example `a := b := 3` is invalid, because on the right hand side of the first `:=` must be an expression. But per definition in XièXie `b := 3` is not an expression, but a statement! The variable list assignment (`a, b, c := ...`) is a comfort functionality, which is only supported for the copy assignment.

The modify-assign operators are available as well:

```
a += b
a -= b
a *= b
a /= b
a %= b
a <<= b
a >>= b
a &= b
a |= b
a ^= b
```

However, they are also not allowed inside another expression.

2.3 Type Denoters

2.3.1 Built-in Types

There are only the following three built-in data types:

```
bool  // Boolean type; can be 'true' or 'false'
int   // 32-bit signed integral type
float // 32-bit floating-point type
```

2.3.2 Objects

Types for class objects (more about classes in chapter 3) are written as follows:

```
// Empty string
String s

// List of strings
String s1 := "Hello, World", s2 := "Foo", s3 := "Bar"
```

2.3.3 Arrays

The only generic way for lists are the built-in arrays:

```
// Declare array objects with initializer lists
int[] intArray := { 1, 2, 3 }
float[][] floatArrayArray := { { 0.0, 1.5 }, { 3.5, 1.23 } }
String[] stringArray := { "a", "b", "c" }

// Access array elements
String s1 := stringArray[0]
String s2 := stringArray[intArray[0]]
float[] floatArray := floatArrayArray[1]
```

2.3.4 Automatic Type Deduction

Whereas automatic type deduction in C++11 is a very extensive language feature, in XièXie it can be summarized in this section. There are two keywords for automatic type deduction: `var` and `const`. As the name implies `var` denotes a variable type and `const` denotes a constant type. The latter type is the only way to define constants in XièXie. Here are a few examples:

```
var   i  := 1       // i is from type 'int'
var   f  := 3.5     // f is from type 'float'
var   s  := "."     // s is from type 'String'
var   a  := { 5 }   // a is from type 'array of int'
```

```
var   aa := {       // aa is from type 'array of array of String'
  { "test" },
  { "a", "b" }
}

const ci := 5       // ci is a constnat int with value 5
const cj := ci*2    // cj is a constant int with value 10
const cf := 3.14    // cf is a constant float
const cb := ci > cj // cb is a constant bool with value 'false'
```

2.4 Statements and Expressions

Unlike C++ and Java, there are no semicolons in XièXie to terminate statements. Only the regular `for`-statement has two semicolons, to separate the initializer statement, the conidition expression, and the increment statement. This means the compiler (or rather the *parser*, which reads the source code) always knows when a statement or expression is complete. But it also means that you — the programmer — can do weird things with this syntax. Consider the following code sample:

```
int a:=3
-4
,b:=-
5+2 int c
```

This is valid XièXie code and it contains only two statements! If we write it in a more common convention, it may look like this:

```
int a := 3-4,
    b := -5+2
int c
```

Hence, the readability of your code is up to you and your programming style :-). The only language I've worked with, which forces you to practice better readability is Python. Actually a great principle, but with XièXie you have complete freedom.

The absence of statement terminators is the reason for the *double paren* syntax of attributes:

```
[[attribute]]
```

To understand why this is the case, let's assume attributes are written with a single paren and consider the following class declaration:

```
class Widget {
    const c := 0 // Initialize member constant 'c' with 0
    int   v := c // Initialize member variable 'v' with constant 'c'
    [entry]      // Mark next procedure as the main entry point with attribute 'entry'
    static void main() {}
}
```

Now this doesn't seem very complex. But the parser runs into trouble when reading `[entry]`. This is because the parser reads it as follows:

```
int v := c[entry]
```

But `c` is not an array. This is why attributes are written with a double paren, because array accesses never begin with '[[‘. They may end with ']]', but this is not important for the parser.

2.4.1 Loop Statements

It follows several examples of loop statements.

for Loop

```
// This regular for loop iterates 'i' from 0 to 9 (similar to Java)
for int i := 0 ; i < 10 ; i++ {
    // Infinite loop (also similar to Java)
    for ;; {
        if i >= 0 {
            // Break infinite loop
            break
        }
```

```
    }

    // Inner iteration variable 'j' is implicit initialized to 0.0
    float j
    for ; j < 3.5 ; {
        j += 0.5
    }
}
```

Ranged Based for Loop

```
// Print numbers 0, 1, 2, 3, and 4
for i : 0 .. 4 {
    // Print value of 'i' (this 'i' is not mutable!)
    print(i)
}

// Print numbers 10, 7, 4, 1, -2, -5, and -8
for i : 10 .. -10 -> 3 {
    print(i)
}

// Do something 10 times (with invisible index variable)
for 10 {
    doSomething()
}
```

foreach Loop

```
// Iterate over array with elements 1, 2, and 3
foreach i : { 1, 2, 3 } {
    print(i)
}

// Iterate over a 'superList' from type (array of array of strings)
String[][] superList
foreach list : superList {
    // Iterate over all elements in the current sub list
    foreach str : list {
        // Do something with this string
        print(str)
    }
}
```

forever Loop

```
// Infinite loop
forever {
    // Condition to break the loop
    if magicFunction() {
        // Break infinite loop
        break
    }
}
```

while Loop

```
// Regular while loop
while magicFunction() {
    doSomething()
}
```

do/while Loop

```
// Regular do-while loop
do {
    doSomething()
} while magicFunction()
```

Chapter 3

Classes

A XièXie program can only consist of imports, modules, and class declarations. And classes can only be defined in the global scope. That means every procedure must be defined inside a class and inner classes are currently not supported.

3.1 Getting Started

To get started, take a look at the following example program which prints the classical phrase "Hello, World!" onto the standard output:

```
1 // XieXie Hello World Program
2 import System
3 class HelloWorld {
4     [[entry]]
5     static void main() {
6         System.out.writeLine("Hello, World!")
7     }
8 }
```

This merely writes the line "`Hello, World!`" to the standard output. Let's take a closer look at each line.

Line 2 imports the "System.xx" file from the XièXie standard library:

```
1 import System      // either this ...
2 import "System.xx" // ... or this
```

If the imported file is in another directory, the string version of `import` is the only choice. This can also be omitted if the file is added to the compilation process. The files for the classes `Object`, `String`, `Array`, and `Intrinsics` are always implicit imported, because they are 'internal' classes the compiler knows generally. Note that verbatim strings are allowed wherever string literals are allowed, i.e. the following example is valid XièXie code:

```
1 import "C:\\Program Files\\Test1.xx" // either this ...
2 import @"C:\Program Files\Test1.xx"  // ... or this
```

The `import` keyword is different to that in Java and also different to the `#include` directive in C++. Although it takes a filename as parameter (like C++'s `#include`), it does not *include* the file in place. Whenever an `import` is read by the *parser*, the filename is added to the set of import files. After all source files have been read, which were passed as input to the compiler, all import files will be read next. This will be repeated until no new files are added to the set. Consider this is a *set* of files, i.e. several `import` commands may occur with the same filename, but it will be read only once. This is why the above sample is valid XièXie code. It also means that recursive imports are allowed as well:

```
1 // File1.xx
2 import "File2.xx"
```

```
1 // File2.xx
2 import "File1.xx"
```

Line 3 declares the class `HelloWorld` which implicit inherits from the base class `Object`, like it is done in Java:

```
1 class HelloWorld { /* ... */ }
```

To inherit from other classes, just write a colon and the identifier of the base class:

```
1 class SubClass : BaseClass { /* ... */ }
```

There is no multiple inheritance like in C++ or interfaces like in JAVA!

The next two lines declare the procedure `main`. In line 4, in *attribute* is defined for this procedure. This makes the procedure to the main **entry point**:

```
1 [[entry]]
2 static void main() { /* ... */ }
```

There are several attributes for class-, procedure-, or variable declarations. Currently only two attributes are supported:

```
1  // class A is marked as 'deprecated'
2  [[deprecated]]
3  class A {}
4
5  // class B is marked as 'deprecated' with a hint
6  [[deprecated("hint...")]]
7  class B {
8      // procedure M1 is marked as the main entry point
9      [[entry]]
10     static void M1() {}
11
12     // procedure M2 with return type is marked as an
13     // alternative entry point named "entryAlt1"
14     [[entry("entryAlt1")]]
15     static int M2() { return 0 }
16
17     // procedure M3 with arguments 'args' is marked as
18     // an alternative entry point named "entryAlt2"
19     [[entry("entryAlt2")]]
20     static void M3(String[] args) {}
21 }
```

The last line of code prints the message to the standard output:

```
1 System.out.writeLine("Hello, World!")
```

`System` is a class from the standard XIÈXIE library, `out` is a *static* member from the type 'OutputStream', and `writeLine` is a function which takes a string as input.

3.2 Declaration Rules for Classes

In XIÈXIE there is no need for *forward declarations*. Everything can be declared in the respective scope and is accessible throughout the entire program (except private scope). This is why the following code is valid:

```
1 // First declare sub class
2 class SubClass : BaseClass { /* ... */ }
3
4 // Then declare base class
5 class BaseClass { /* ... */ }
```

The same applies for procedure declarations:

```
1  class B {
2      static void procB1() {
3          A.procA()  // no forward declaration required ...
4          B.procB2() // ... same here
5      }
6      static void procB2() { /* ... */ }
7  }
8  class A {
9      static void procA() { /* ... */ }
10 }
```

This works because the *context analyzer* of the compiler works in several phases:

1. Class symbols are registered in global scope.
2. Class signatures are analyzed (attributes and base class).
3. Class inheritance is verified (check for cycles).

4. Class member symbols are registred in respective class scope (procedures, variables, etc.).
5. Procedure code is analyzed.

3.3 Procedures

In XièXie we talk about *procedures* (somethings called *methods*), because in the strict sense *functions* have no side effects. Functions have only input parameters which are calculated to a result. But in XièXie every procedure can have side effects, meaning that they can modify the program state (with static variables for instance).

3.3.1 Procedure Signature

A *procedure signature* is the extended identification of a procedure beyond its identifier string. The signature consists of the identifier string, its parameter list, and the procedure's return type. However, the return type is never used for identification.

3.3.2 Procedure Overloading

XièXie supports overloading of procedures. This means the same identifier can be used several times for procedures inside a class declaration (including its inheritance hierarchy). This requires that the following rules are satisfied:

> All procedures with the same identifier inside a class declaration can be distinguished by their parameter count or parameter types.

Here is an example of procedure overloading.

```
class Widget {
    int f() {
        return 1
    }
    int f(int x) {
        return x*2
    }
    float f(float x) {
        return x*3.0
    }
    void caller() {
        int   a := f()    // a is 1
        int   b := f(1)   // b is 2
        float c := f(2.0) // c is 6.0
    }
}
```

When overloading procedures, try to avoid ambiguities with default arguments. Adding default arguments to all parameters of all overloaded procedures is allowed, but procedure calls may be ambiguous for the compiler:

```
class Widget {
    int f() { /* ... */ }
    int f(int x := 0) { /* ... */ }
    float f(float x := 0.0) { /* ... */ }

    void caller() {
        int   a := f()    // error, could be f(), f(int x := 0), or f(float x := 0.0)
        int   b := f(1)   // ok, argument is from type 'int'
        float c := f(2.0) // ok, argument is from type 'float'
    }
}
```

3.3.3 Procedure Overriding

XièXie supports overriding of procedures. This means the same procedure signature can be used inside a class and its base class. The procedure calls of overloaded and overridden procedures require that the following rules are satisfied:

> If a procedure with identifier *P* is declared inside a class *C*, a procedure with the same identifier is declared in its base class *B* but with another signature, and another procedure inside class *C* calls the procedure *P* from class *B*, then the identifier `super` must be specified in front of the call.

To better understand this awkward definition, take a look at this example:

```
class B {
    int f(int x, int y) {
        return 1
    }
    int g() {
        return 2
    }
}
class S : B {
    int f(int x) {
        return 3
    }
    void caller() {
        int a := f(0)          // equivalent to 'this.f(0)'
        int b := super.f(0, 0) // 'super' is required, due to overloaded procedure 'f'
        int c := g()           // no need for 'super', because 'g' is not overloaded
    }
}
```

Chapter 4

Modules

A *module* consists of a shared library (`*.dll` on Windows and `*.so` on GNU/Linux) and a XièXie module file. They are similar to Python modules.

4.1 Using Modules

Module declarations are similar to external classes. But they can only have static procedures. Here is an example:

```
module MyModule {
    static void doSomething(int x)
}
```

Using this in your XièXie code will then look like this:

```
import MyModule
class MyClass {
    static void main() {
        MyModule.doSomething(42)
    }
}
```

4.2 Creating Modules

Modules must be written in plain C or C++. A minimal module would consist of three files. Supposed our example module is named "MyModule" and we work on Windows we would have these files:

- **`MyModule.c`** Module code written in C.
- **`MyModule.dll`/ `.so`** Shared library which runs the module code.
- **`MyModule.xx`** XièXie code file which declares the module.

The C code file must implement the following function interfaces to be a valid XièXie module:

```
// Returns the number of module procedures.
int xx_module_proc_count();

// Returns the procedure for the respective index.
XVM_INVOCATION_PROC xx_module_fetch_proc(int index);

// Returns the procedure identifier for the respective index.
const char* xx_module_fetch_ident(int index);
```

Our simple module example from above could be implemented as follows:

```
#include "../xx_module.h"

// This is our module procedure "soSomething".
// All module procedures must have this interface.
void doSomething(xvm_env env) {
    // Get 1st parameter from XVM environment
    int x = xvm_param_int(env, 1);

```

```
    // Do something with 'x'
    printf("input parameter 'x' is %i\n", i);

    // Pop arguments from stack (1 for the single parameter 'x')
    xvm_return_void(env, 1);
}

// Implement the module interface
static xvm_invocation procList[] = {
    { "doSomething", doSomething },
};

XVM_IMPLEMENT_MODULE_INTERFACE(procList);
```

Part II

Developer Tools

Chapter 5

Compiler

5.1 Command Line Tool

The compiler can be used as command line tool. The appropriate program is named `xxc`. Enter `xxc help` in a command line to see the manual pages.

In contrast to most other command line tools, the commands for `xxc` don't have the '-' prefix. Instead the command options use this prefix. Here is an example:

```
xxc C -f FILE1 -f FILE2 -O
```

The above command line uses the `C` command (also `compile`) with the flags '-f' and '-O'.

5.1.1 Output

The compiler tries to show you where an error or warning occurred. If possible, it is highlighted with a *line marker*. Consider the following code sample:

```
1 class Foo {
2     void Bar() {
3         int x := y
4     }
5 }
```

Now consider the compiler processes this code with the following command line:

```
xxc C -f Core.xx -f Foo.xx
```

Then the compiler indicates that the variable 'y' was not declared (see Figure 5.1). As you can see, some errors may produce multiple outputs. In this case, 'y' is undeclared which produces two error outputs here:

1. The identifier 'y' is undeclared, so the appearance in the initializer expression is invalid.
2. The type of the initializer expression for the declaration of 'x' can not be deduced, so the type check fails.



(a) Output in PowerShell on Windows 7



(b) Output in Terminal on Kubuntu GNU/Linux

Figure 5.1: Compiler output for the above code sample.

Chapter 6

Virtual Machine

The XièXie Virtual Machine (XVM) is a separated program (named `xvm`), written in pure C99.

6.1 Executing Programs

To execute (or run) a virtual program, just enter the filename of an `*.xbc` file into the `xvm`:

```
xvm HelloWorld.xbc
```

Part III

Low-Level Programming

Chapter 7

Virtual Assembler

The XièXie Virtual Assembler (XASM) is the low level language which is used as interface to the `xvm`. It's a very low level language, with similarities to ARM® Assembler.

7.1 Registers

The `xvm` is a register machine, i.e. it uses registers as operands for its instructions instead of the stack. This commonly increases performance but is a little more tricky to use, when you run out of registers. Each register index inside an instruction is stored with 4 bits, thus there are 16 registers and this is the list:

1. `$r0` General Purpose Register 0.

   ⋮

10. `$r9` General Purpose Register 9.
11. `$tr` Temporary Register.
12. `$gp` Global Pointer.
13. `$cf` Conditional Flag.
14. `$lb` Local Base Pointer.
15. `$sp` Stack Pointer.
16. `$pc` Program Counter.

All pointer registers and the program counter (i.e. `$gp`, `$lb`, `$sp`, and `$pc`) are all 'real' pointers to the memory in your operating system.

7.2 Instruction Set

Each instruction stores its mnemonic (Greek 'memory' ≙ instruction identifier) in the first 6 bits. This provides a maximum of 64 instructions. However, only 52 mnemonics are currently in use.

3-Register Instruction OpCodes (00....)						
Mnemonic	OpCode 31.....26	Dest. 25...22	LSource 21...18	RSource 17...14	Unused 13.....0	Description
AND	0 0 0 0 0 1	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Bitwise AND (D := L & R).
OR	0 0 0 0 1 0	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Bitwise OR (D := L | R).
XOR	0 0 0 0 1 1	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Bitwise XOR (D := L ^ R).
ADD	0 0 0 1 0 0	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Integer add. (D := L + R).
SUB	0 0 0 1 0 1	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Integer sub. (D := L - R).
MUL	0 0 0 1 1 0	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Integer mul. (D := L × R).
DIV	0 0 0 1 1 1	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Integer div. (D := L ÷ R).
MOD	0 0 1 0 0 0	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Modulo (D := L % R).
SLL	0 0 1 0 0 1	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Shift left (D := L « R).
SLR	0 0 1 0 1 0	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Shift right (D := L » R).
ADDF	0 0 1 0 1 1	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Float add. (D := L + R).
SUBF	0 0 1 1 0 0	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Float sub. (D := L - R).
MULF	0 0 1 1 0 1	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Float mul. (D := L × R).
DIVF	0 0 1 1 1 0	D D D D	L L L L	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0	Float div. (D := L ÷ R).

2-Register Instruction OpCodes (01....)

Mnemonic	OpCode 31.....26	Dest. 25...22	Source 21...18	Unused or Value 17.....0	Description
MOV	0 1 0 0 0 0	D D D D	S S S S	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Move (D := S).
NOT	0 1 0 0 0 1	D D D D	S S S S	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Bitwise NOT (D := ~S).
FTI	0 1 0 0 1 0	D D D D	S S S S	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Float to integer (D := (int)S).
ITF	0 1 0 0 1 1	D D D D	S S S S	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Integer to float (D := (float)S).
AND	0 1 0 1 0 0	D D D D	S S S S	V V V V V V V V V V V V V V V V V V	Bitwise AND (D := S & V).
OR	0 1 0 1 0 1	D D D D	S S S S	V V V V V V V V V V V V V V V V V V	Bitwise OR (D := S | V).
XOR	0 1 0 1 1 0	D D D D	S S S S	V V V V V V V V V V V V V V V V V V	Bitwise XOR (D := S ^ V).
ADD	0 1 0 1 1 1	D D D D	S S S S	V V V V V V V V V V V V V V V V V V	Modulo (D := S % V).
SUB	0 1 1 0 0 0	D D D D	S S S S	V V V V V V V V V V V V V V V V V V	Shift left (D := S « V).
MUL	0 1 1 0 0 1	D D D D	S S S S	V V V V V V V V V V V V V V V V V V	Shift right (D := S » V).
DIV	0 1 1 0 1 0	D D D D	S S S S	V V V V V V V V V V V V V V V V V V	Integer add. (D := S + V).
MOD	0 1 1 0 1 1	D D D D	S S S S	V V V V V V V V V V V V V V V V V V	Integer sub. (D := S - V).
SLL	0 1 1 1 0 0	D D D D	S S S S	V V V V V V V V V V V V V V V V V V	Integer mul. (D := S × V).
SLR	0 1 1 1 0 1	D D D D	S S S S	V V V V V V V V V V V V V V V V V V	Integer div. (D := S ÷ V).
CMP	0 1 1 1 1 0	X X X X	Y Y Y Y	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Integer comparison (→ $cf).
CMPF	0 1 1 1 1 1	X X X X	Y Y Y Y	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Float comparison (→ $cf).

1-Register Instruction OpCodes (100...)

Mnemonic	OpCode 31.....26	Reg. 25...22	Unused, Value, or Offset 21.....0	Description
PUSH	1 0 0 0 0 0	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Push register onto stack.
POP	1 0 0 0 0 1	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Pop register from stack.
INC	1 0 0 0 1 0	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Increment integer (D++).
DEC	1 0 0 0 1 1	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Decrement integer (D–).
MOV	1 0 0 1 0 0	R R R R	V V V V V V V V V V V V V V V V V V V V V V	Move integer (D := V).
LDA	1 0 0 1 0 1	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Load address from program (D := ByteCode + R_{word} + O_{byte}).

Jump Instruction OpCodes (101...)

Mnemonic	OpCode 31.....26	Reg. 25...22	Offset 21.....0	Description
JMP	1 0 1 0 0 0	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump.
JE	1 0 1 0 0 1	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if equal.
JNE	1 0 1 0 1 0	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if not-equal.
JG	1 0 1 0 1 1	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if greater.
JL	1 0 1 1 0 0	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if less.
JGE	1 0 1 1 0 1	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if greater or equal.
JLE	1 0 1 1 1 0	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Jump if less or equal.
CALL	1 0 1 1 1 1	R R R R	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Push dynamic link ($lb and $pc) onto stack; Set $lb to new stack frame; Jump to address.

Load/Store Instruction OpCodes (1100..)

Mnemonic	OpCode 31.....26	Reg. 25...22	Addr. 21...18	Offset 17.....0	Description
LDB	1 1 0 0 0 0	R R R R	A A A A	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Load byte from memory.
STB	1 1 0 0 0 1	R R R R	A A A A	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Store byte to memory.
LDW	1 1 0 0 1 0	R R R R	A A A A	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Load word from memory.
STW	1 1 0 0 1 1	R R R R	A A A A	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Store word to memory.

Special-1 Instruction OpCodes

Mnemonic	OpCode 31.....26	Unused or Value 25.....0	Description
STOP	0 0 0 0 0 0	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0	Stop program execution.
PUSH	1 1 1 0 0 0	V V V V V V V V V V V V V V V V V V V V V V V V V V	Push integer onto stack.
INVK	1 1 1 0 0 1	V V V V V V V V V V V V V V V V V V V V V V V V V V	Invoke external procedure.

Special-2 Instruction OpCodes

Mnemonic	OpCode 31.....26	Result Size 25.....18	Argument Size 17.....0	Description
RET	1 1 1 0 1 0	R R R R R R R R	A A A A A A A A A A A A A A A A A A	Pop R wrods from stack and buffer them; Pop current stack frame; Pop A words from stack; Push stored R words back onto stack; Restore dynamic link ($lb and $pc)

7.3 Calling Convention

In XASM, the calling convention provides that the *callee* removes the procedure arguments from the stack. After any procedure call, the *dynamic link* is stored at the beginning of the *stack frame* and reserves the first 8 bytes.

dynamic link

Consists of two words: 32-bit value of the `$lb` register *before* the procedure call and 32-bit value of the `$pc` register *before* the procedure call.

Local variables can be store by increasing the *stack pointer* (`$sp`) and load/store instructions at the *local base pointer* (`$lb`) plus 8 bytes (after the *dynamic link*). Here is an example:

```
1  ; void proc() { ... }
2  proc:
3    ADD $sp, $sp, 8    ; Allocate memory on the stack for two words (2*4 bytes)
4    MOV $r0, 42        ; Store value 42 in register $r0
5    STW $r0, ($lb) 8   ; Store register $r0 in local scope at first position (after dynamic link)
6    ADD $r0, $r0, 10   ; Increment value of register $r0
7    STW $r0, ($lb) 12  ; Store register $r0 in local scope at second position
8    ...
```

The arguments for a procedure call are meant to be pushed onto the stack from *right-to-left*. In this way the arguments can be accessed from *left-to-right*. Here is an example:

```
1  ; int proc(int x, int y) { return x+y }
2  proc:
3    LDW $r0, ($lb) -4 ; Fetch first argument (x) from stack
4    LDW $r1, ($lb) -8 ; Fetch second argument (y) from stack
5    ADD $r0, $r0, $r1 ; Calculate x := x + y
6    PUSH $r0          ; Push result onto stack
7    RET (1) 2         ; Return: result size = 1 word, argument size = 2 words
```