

# XièXiè Programming Guide

Version 2.0 Alpha

Lukas HERMANNs

March 18, 2015

# About the Author

My name is Lukas Hermanns (age-group 1990) and I started this project during my studies in 2014. By now I have over 12 years of experience in computer programming, started at the age of 12. I have been writing programs in Basic languages such as QBASIC, PUREBASIC, and BLITZ3D; in high level languages such as C, C++, C#, OBJECTIVE-C, and JAVA; but also in scripting languages such as JAVASCRIPT and PYTHON. I'm actually a preferred programmer in C++ (meanwhile C++11), low level stuff, and graphics programming with shading languages such as GLSL and HLSL.

The XiEXiE programming language is intended to be simple and not tuned for performance. It was originally designed to be used for scripting in video games, but can also be used for general purposes.

If you like, you can follow me on [Twitter](#), [YouTube](#), [GitHub](#), or [Bitbucket](#).

# ToDo List

The compiler, and partially the virtual machine, are not yet completed. Some rules and explanations in this report may change over time. Here is a rough ToDo-list:

PARSER	
Implementation of parser against grammar specification	almost done
Member procedure calls as statements	not yet implemented
Post fix array access	done
Destructor	not yet implemented
Named Parameters	done
CONTEXT ANALYZER	
Expression Type Check	done
Cast Type Check	incomplete
Automatic Type Deduction	done
Procedure Overloading	done
Procedure Overriding	done
Named Parameters	not yet implemented
Static and Non-Static Procedure Behavior	incomplete
Class and Procedure Attributes	not yet implemented
CODE GENERATOR	
CFG and TAC generation from DAST	not yet implemented
XASM BACKEND	
Final code generation for XASM	not yet implemented

# Contents

<b>I</b>	<b>The XièXiè Programming Language</b>	<b>4</b>
<b>1</b>	<b>Introduction</b>	<b>5</b>
1.1	What is XièXiè?	5
1.2	Why is it called “XièXiè”?	5
1.3	Motivation	5
<b>2</b>	<b>Syntax</b>	<b>6</b>
2.1	Basics	6
2.1.1	Commentaries	6
2.1.2	Identifiers	6
2.1.3	Literals	7
2.2	Operators	7
2.3	Type Denoters	8
2.3.1	Built-in Types	8
2.3.2	Objects	8
2.3.3	Arrays	8
2.3.4	Automatic Type Deduction	8
2.4	Statements and Expressions	9
2.4.1	Loop Statements	9
<b>3</b>	<b>Classes</b>	<b>12</b>
3.1	Getting Started	12
3.2	Declaration Rules for Classes	13
3.3	Procedures	14
3.3.1	Procedure Signature	14
3.3.2	Procedure Overloading	14
3.3.3	Procedure Overriding	14
<b>II</b>	<b>Developer Tools</b>	<b>16</b>
<b>4</b>	<b>Compiler</b>	<b>17</b>
4.1	Command Line Tool	17
<b>5</b>	<b>Virtual Machine</b>	<b>18</b>
5.1	Executing Programs	18

## **Part I**

# **The XIÈXIE Programming Language**

# Chapter 1

## Introduction

First of all, “xièxie” is the chinese word for “thanks” or “thank you” (see [hantrainerpro.com](http://hantrainerpro.com)) and is roughly pronounced “Sh-eh sh-eh”.

The design of the XièXie programming language is overall influenced by `JAVA`, `C++`, and `PYTHON`.

### 1.1 What is XièXie?

The XièXie programming language is a high-level, object-oriented, scripting language with compiler and virtual machine. The XièXie compiler (XXC) translates the XièXie code (XX) to a virtual assembler (XASM), then assembles it to XièXie byte code (XBC), which can then be interpreted by the XièXie virtual machine (XVM).

### 1.2 Why is it called “XièXie”?

Some years, before I started with the development of this compiler, I already had some ideas about a name for it — at this time the compiler was intended to be a cross compiler, i.e. to compile the XièXie code to `C++`. The first idea I had in mind, was to call it “C+=2” or “C++++” because it should be a more comfortable and easier `C++` variant. But this name looked and sounded strange. Another idea was to call it “C power of C” (in mathematical notation  $C^C$ ). But that still was not what I was looking for. Then I remembered me to the Chinese word “Xièxie” which means “Thanks” in English and it sounds a little similar to “CC” which could be seen as a shortcut for  $C^C$ . That’s the XièXie compiler’s story about its name.

Or to make a long story short: I saw this word on a napkin in a Chinese restaurant and thought to my self: That’s it! :-)

### 1.3 Motivation

There are many great programming languages out there. Some produce faster code than others, but some are simpler and have a better learning curve. Using a new language doesn’t mean to give up a previous one, because every language has its own domain. For example, the performance of interpreted languages such as `PYTHON` is totally sufficient for many applications. We don’t need maximal performance for a script which does some socket connection or text processing for instance. But I would not write the compiler and interpreter for `PYTHON` in a scripted language. There are also many ways to combine several languages: an interpreter could be used for scripting in video games for instance. But the game engine is written in `C++`. This is how it is done in `UNITY3D` (see [www.unity3d.com](http://www.unity3d.com)). They use `MONO` as compiler and interpreter framework for `C#`. But the engine itself is written in `C++`.

Now XièXie is aimed to be used for small scripting purposes, with a gentle learning curve. The great thing about its interpreter is, that it is very tiny and can easily be integrated into existing `C` or `C++` code. This virtual machine only consists of a single code file (written in `C99`) and can simply be included into any `C99` compliant project.

# Chapter 2

## Syntax

We start out with the syntax.

### 2.1 Basics

#### 2.1.1 Commentaries

Commentaries are a fundamental part of programming languages and they are nearly identical to those in `JAVA`:

```
1 // Single-line comment
2
3 /* Single-line comment */
4
5 /*
6 Multi-line comment
7 */
```

Although they are very similar to the commentaries in `JAVA`, nested multi-line comments are allowed as well:

```
1 /*
2 Outer comment
3 /* Nested comment */
4 */
```

#### 2.1.2 Identifiers

Identifiers (for variables, classes, etc.) must only contain alpha-numeric characters and the underscore. They must also begin with a letter or an underscore. But there is a small exception: they must not begin with `__xx__`, because all identifiers with this prefix are reserved for internal use of the compiler.

Valid identifiers are:

- `_name_`
- `FooBar`
- `number_of_wheels`
- `Customer01`
- ...

Invalid identifiers are:

- `naïve`
- `Foo.Bar`
- `number-of-wheels`
- `3over2`
- ...

### 2.1.3 Literals

These are the kinds of literals:

- Boolean Literal: `true`, `false`
- Integer Literal (Binary): e.g. `0b11001`, `0b0000`
- Integer Literal (Octal): e.g. `0o24`, `0o01234567`
- Integer Literal (Decimal): e.g. `3`, `12`, `999`, `1234567890`
- Integer Literal (Hexa-Decimal): e.g. `0xff`, `0x00`, `0xaB29`
- Float Literal: e.g. `0.0`, `3.5`, `12.482`
- String Literal: e.g. `"Foo Bar"`, `"Hello, World"`, `"\n\t"`, `"1st fragment"_"2nd fragment"`
- Verbatim String Literal: e.g. `@"\home\test"`, `@"a ""b"" c"`

Integer and float literals may optionally contain the single quotation mark as digit separator, for better readability. If it's used, all separators must satisfy the following rules:

1. For decimal literals, the separators must be **three steps apart** from each other, beginning at the dot.  
Example: `12'345`, `3.141'592'654`, or `-27'836.283'74`.
2. For non-decimal literals, the separators must be **four steps apart** from each other, beginning at the dot.  
Example: `0xff0'214b`, `0b100'1101'1011`, `0o1234'5670`
3. A separator must not appear at the beginning or the end of the literal.  
Counterexample: `'123'456'`.
4. No valid separator must be omitted.  
Counterexample: `1234'567'890`.

## 2.2 Operators

The most operators as in JAVA or C++ are also available in XIE:

```
1  /* Arithmetic Operators */
2  a + b    // Addition
3  a - b    // Substration
4  a * b    // Multiplication
5  a / b    // Division
6  a % b    // Modulo
7  a << b   // Left Shift
8  a >> b   // Right Shift
9  -a      // Negate
10
11 /* Bitwise Operators */
12 a & b    // Bitwise AND
13 a | b    // Bitwise OR
14 a ^ b    // Bitwise XOR
15 ~a      // Bitwise NOT
16
17 /* Boolean Operators */
18 not a    // Logic NOT
19 a and b  // Logic AND
20 a or b   // Logic OR
21
22 /* Relation Operators */
23 a = b    // Equality
24 a != b   // Inequality
25 a < b    // Less
26 a <= b   // Less Or Equal
27 a > b    // Greater
28 a >= b   // Greater Or Equal
```

As the interested reader may have noticed, the *equality operator* is different to that in the most languages. This is because the *copy assignment operators* is `:=` and not `=`.



```

1 x := 5           // ok, set x to 5
2 a, b, c := 4     // ok, set a, b, and c to 4
3 a := b := 3     // error, b := 3 is not an expression

```

In the above example `a := b := 3` is invalid, because on the right hand side of the first `:=` must be an expression. But per definition in XiEXIE `b := 3` is not an expression, but a statement! The variable list assignment (`a, b, c := ...`) is a comfort functionality, which is only supported for the copy assignment.

The modify-assign operators are available as well:

```

1 a += b
2 a -= b
3 a *= b
4 a /= b
5 a %= b
6 a <<= b
7 a >>= b
8 a &= b
9 a |= b
10 a ^= b

```

However, they are also not allowed inside another expression.

## 2.3 Type Denoters

### 2.3.1 Built-in Types

There are only the following three built-in data types:

```

1 bool // Boolean type; can be 'true' or 'false'
2 int  // 32-bit signed integral type
3 float // 32-bit floating-point type

```

### 2.3.2 Objects

Types for class objects (more about classes in chapter 3) are written as follows:

```

1 // Empty string
2 String s
3
4 // List of strings
5 String s1 := "Hello, World", s2 := "Foo", s3 := "Bar"

```

### 2.3.3 Arrays

The only generic way for lists are the built-in arrays:

```

1 // Declare array objects with initializer lists
2 int[] intArray := { 1, 2, 3 }
3 float[][] floatArray := { { 0.0, 1.5 }, { 3.5, 1.23 } }
4 String[] stringArray := { "a", "b", "c" }
5
6 // Access array elements
7 String s1 := stringArray[0]
8 String s2 := stringArray[intArray[0]]
9 float[] floatArray := floatArrayArray[1]

```

### 2.3.4 Automatic Type Deduction

Whereas automatic type deduction in C++11 is a very extensive language feature, in XiEXIE it can be summarized in this section. There are two keywords for automatic type deduction: `var` and `const`. As the name implies `var` denotes a variable type and `const` denotes a constant type. The latter type is the only way to define constants in XiEXIE. Here are a few examples:

```

1 var i := 1           // i is from type 'int'
2 var f := 3.5         // f is from type 'float'
3 var s := "."         // s is from type 'String'
4 var a := { 5 }       // a is from type 'array of int'

```

```

5 var aa := {           // aa is from type 'array of array of String'
6   { "test" },
7   { "a", "b" }
8 }
9
10 const ci := 5         // ci is a constnat int with value 5
11 const cj := ci*2       // cj is a constant int with value 10
12 const cf := 3.14       // cf is a constant float
13 const cb := ci > cj    // cb is a constant bool with value 'false'

```

## 2.4 Statements and Expressions

Unlike C++ and JAVA, there are no semicolons in XiEXIE to terminate statements. Only the regular `for`-statement has two semicolons, to separate the initializer statement, the condition expression, and the increment statement. This means the compiler (or rather the *parser*, which reads the source code) always knows when a statement or expression is complete. But it also means that you — the programmer — can do weird things with this syntax. Consider the following code sample:

```

1 int a:=3
2 -4
3 ,b:=-
4 5+2 int c

```

This is valid XiEXIE code and it contains only two statements! If we write it in a more common convention, it may look like this:

```

1 int a := 3-4,
2     b := -5+2
3 int c

```

Hence, the readability of your code is up to you and your programming style :-). The only language I've worked with, which forces you to practice better readability is PYTHON. Actually a great principle, but with XiEXIE you have complete freedom.

The absence of statement terminators is the reason for the *double paren* syntax of attributes:

```

1 [[attribute]]

```

To understand why this is the case, let's assume attributes are written with a single paren and consider the following class declaration:

```

1 class Widget {
2     const c := 0 // Initialize member constant 'c' with 0
3     int v := c // Initialize member variable 'v' with constant 'c'
4     [entry] // Mark next procedure as the main entry point with attribute 'entry'
5     static void main() {}
6 }

```

Now this doesn't seem very complex. But the parser runs into trouble when reading `[entry]`. This is because the parser reads it as follows:

```

1 int v := c[entry]

```

But `c` is not an array. This is why attributes are written with a double paren, because array accesses never begin with '['. They may end with ']', but this is not important for the parser.

### 2.4.1 Loop Statements

It follows several examples of loop statements.

#### for Loop

```

1 // This regular for loop iterates 'i' from 0 to 9 (similar to Java)
2 for int i := 0 ; i < 10 ; i++ {
3     // Infinite loop (also similar to Java)
4     for ;; {
5         if i >= 0 {
6             // Break infinite loop
7             break
8         }

```

```

9     }
10
11     // Inner iteration variable 'j' is implicit initialized to 0.0
12     float j
13     for ; j < 3.5 ; {
14         j += 0.5
15     }
16 }

```

## Ranged Based for Loop

```

1 // Print numbers 0, 1, 2, 3, and 4
2 for i : 0 .. 4 {
3     // Print value of 'i' (this 'i' is not mutable!)
4     print(i)
5 }
6
7 // Print numbers 10, 7, 4, 1, -2, -5, and -8
8 for i : 10 .. -10 -> 3 {
9     print(i)
10 }
11
12 // Do something 10 times (with invisible index variable)
13 for 10 {
14     doSomething()
15 }

```

## foreach Loop

```

1 // Iterate over array with elements 1, 2, and 3
2 foreach i : { 1, 2, 3 } {
3     print(i)
4 }
5
6 // Iterate over a 'superList' from type (array of array of strings)
7 String[][] superList
8 foreach list : superList {
9     // Iterate over all elements in the current sub list
10    foreach str : list {
11        // Do something with this string
12        print(str)
13    }
14 }

```

## forever Loop

```

1 // Infinite loop
2 forever {
3     // Condition to break the loop
4     if magicFunction() {
5         // Break infinite loop
6         break
7     }
8 }

```

## while Loop

```

1 // Regular while loop
2 while magicFunction() {
3     doSomething()
4 }

```

## do/while Loop

```
1 // Regular do-while loop
2 do {
3     doSomething()
4 } while magicFunction()
```

# Chapter 3

## Classes

A XIEIE program can only consist of imports and class declarations. And classes can only be defined in the global scope. That means every procedure must be defined inside a class and inner classes are currently not supported.

### 3.1 Getting Started

To get started, take a look at the following example program which prints the classical phrase “Hello, World!” onto the standard output:

```
1 // XieXie Hello World Program
2 import System
3 class HelloWorld {
4     [[entry]]
5     static void main() {
6         System.out.WriteLine("Hello, World!")
7     }
8 }
```

This merely writes the line “Hello, World!” to the standard output. Let’s take a closer look at each line. Line 2 imports the “System.xx” file from the XIEIE standard library:

```
1 import System // either this ...
2 import "System.xx" // ... or this
```

If the imported file is in another directory, the string version of import is the only choice. This can also be omitted if the file is added to the compilation process. The files for the classes Object, String, Array, and Intrinsics are always implicit imported, because they are ‘internal’ classes the compiler knows generally. Note that verbatim strings are allowed wherever string literals are allowed, i.e. the following example is valid XIEIE code:

```
1 import "C:\\Program_Files\\Test1.xx" // either this ...
2 import @"C:\\Program_Files\\Test1.xx" // ... or this
```

The import keyword is different to that in JAVA and also different to the #include directive in C++. Although it takes a filename as parameter (like C++’s #include), it does not *include* the file in place. Whenever an import is read by the *parser*, the filename is added to the set of import files. After all source files have been read, which were passed as input to the compiler, all import files will be read next. This will be repeated until no new files are added to the set. Consider this is a *set* of files, i.e. several import commands may occur with the same filename, but it will be read only once. This is why the above sample is valid XIEIE code. It also means that recursive imports are allowed as well:

```
1 // File1.xx
2 import "File2.xx"
```

```
1 // File2.xx
2 import "File1.xx"
```

Line 3 declares the class HelloWorld which implicit inherits from the base class Object, like it is done in JAVA:

```
1 class HelloWorld { /* ... */ }
```

To inherit from other classes, just write a colon and the identifier of the base class:

```
1 class SubClass : BaseClass { /* ... */ }
```

There is no multiple inheritance like in C++ or interfaces like in JAVA!

The next two lines declare the procedure `main`. In line 4, in *attribute* is defined for this procedure. This makes the procedure to the main **entry point**:

```
1 [[entry]]
2 static void main() { /* ... */ }
```

There are several attributes for class-, procedure-, or variable declarations. Currently only two attributes are supported:

```
1 // class A is marked as 'deprecated'
2 [[deprecated]]
3 class A {}
4
5 // class B is marked as 'deprecated' with a hint
6 [[deprecated("hint...")]]
7 class B {
8     // procedure M1 is marked as the main entry point
9     [[entry]]
10    static void M1() {}
11
12    // procedure M2 with return type is marked as an
13    // alternative entry point named "entryAlt1"
14    [[entry("entryAlt1")]]
15    static int M2() { return 0 }
16
17    // procedure M3 with arguments 'args' is marked as
18    // an alternative entry point named "entryAlt2"
19    [[entry("entryAlt2")]]
20    static void M3(String[] args) {}
21 }
```

The last line of code prints the message to the standard output:

```
1 System.out.writeLine("Hello, World!")
```

`System` is a class from the standard `XiEXIE` library, `out` is a *static* member from the type 'OutputStream', and `writeLine` is a function which takes a string as input.

## 3.2 Declaration Rules for Classes

In `XiEXIE` there is no need for *forward declarations*. Everything can be declared in the respective scope and is accessible throughout the entire program (except private scope). This is why the following code is valid:

```
1 // First declare sub class
2 class SubClass : BaseClass { /* ... */ }
3
4 // Then declare base class
5 class BaseClass { /* ... */ }
```

The same applies for procedure declarations:

```
1 class B {
2     static void procB1() {
3         A.procA() // no forward declaration required ...
4         B.procB2() // ... same here
5     }
6     static void procB2() { /* ... */ }
7 }
8 class A {
9     static void procA() { /* ... */ }
10 }
```

This works because the *context analyzer* of the compiler works in several phases:

1. Class symbols are registered in global scope.
2. Class signatures are analyzed (attributes and base class).
3. Class inheritance is verified (check for cycles).

4. Class member symbols are registered in respective class scope (procedures, variables, etc.).
5. Procedure code is analyzed.

### 3.3 Procedures

In XiEXIE we talk about *procedures* (somethings called *methods*), because in the strict sense *functions* have no side effects. Functions have only input parameters which are calculated to a result. But in XiEXIE every procedure can have side effects, meaning that they can modify the program state (with static variables for instance).

#### 3.3.1 Procedure Signature

A *procedure signature* is the extended identification of a procedure beyond its identifier string. The signature consists of the identifier string, its parameter list, and the procedure's return type. However, the return type is never used for identification.

#### 3.3.2 Procedure Overloading

XiEXIE supports overloading of procedures. This means the same identifier can be used several times for procedures inside a class declaration (including its inheritance hierarchy). This requires that the following rules are satisfied:

All procedures with the same identifier inside a class declaration can be distinguished by their parameter count or parameter types.

Here is an example of procedure overloading.

```

1 class Widget {
2   int f() {
3     return 1
4   }
5   int f(int x) {
6     return x*2
7   }
8   float f(float x) {
9     return x*3.0
10  }
11  void caller() {
12    int a := f()    // a is 1
13    int b := f(1)   // b is 2
14    float c := f(2.0) // c is 6.0
15  }
16 }
```

When overloading procedures, try to avoid ambiguities with default arguments. Adding default arguments to all parameters of all overloaded procedures is allowed, but procedure calls may be ambiguous for the compiler:

```

1 class Widget {
2   int f() { /* ... */ }
3   int f(int x := 0) { /* ... */ }
4   float f(float x := 0.0) { /* ... */ }
5
6   void caller() {
7     int a := f()    // error, could be f(), f(int x := 0), or f(float x := 0.0)
8     int b := f(1)   // ok, argument is from type 'int'
9     float c := f(2.0) // ok, argument is from type 'float'
10  }
11 }
```

#### 3.3.3 Procedure Overriding

XiEXIE supports overriding of procedures. This means the same procedure signature can be used inside a class and its base class. The procedure calls of overloaded and overridden procedures require that the following rules are satisfied:

If a procedure with identifier *P* is declared inside a class *C*, a procedure with the same identifier is declared in its base class *B* but with another signature, and another procedure inside class *C* calls the procedure *P* from class *B*, then the identifier *super* must be specified in front of the call.

To better understand this awkward definition, take a look at this example:

```
1 class B {
2     int f(int x, int y) {
3         return 1
4     }
5     int g() {
6         return 2
7     }
8 }
9 class S : B {
10     int f(int x) {
11         return 3
12     }
13     void caller() {
14         int a := f(0)           // equivalent to 'this.f(0)'
15         int b := super.f(0, 0) // 'super' is required, due to overloaded procedure 'f'
16         int c := g()           // no need for 'super', because 'g' is not overloaded
17     }
18 }
```



# **Part II**

## **Developer Tools**

## Chapter 4

# Compiler

### 4.1 Command Line Tool

The compiler can be used as command line tool. The appropriate program is named `xxc`. Enter `xxc help` in a command line to see the manual pages.

In contrast to most other command line tools, the commands for `xxc` don't have the `'-'` prefix. Instead the command options use this prefix. Here is an example:

```
1 xxc compile -f FILE1 -f FILE2 -O
```

The above command line uses the `compile` command with the flags `'-f'` and `'-O'`.

## Chapter 5

# Virtual Machine

The XièXiè VIRTUAL MACHINE (XVM) is a separated program (named `xvm`), written in pure C99.

### 5.1 Executing Programs

To execute (or run) a virtual program, just enter the filename of an `*.xbc` file into the `xvm`:

```
1 xvm HelloWorld.xbc
```